
Introduction
An individual-based model (sensu DeAngelis and Gross 

1992) of forest stand and landscape dynamics is a useful 
tool for multi-scale analyses of issues in forest ecology and 
management.  Many forest landscape modeling efforts have 
sacrificed information on individual trees for efficient model 
execution.  Recent advances in computational power have 
made it feasible to execute well-designed individual-based 
models over landscapes.  Such models can address problems 
across a broad range of spatial scales from internal stand struc-
ture to landscape patterns and dynamics.

The individual-based landscape model (FOREL) dis-
cussed herein is constructed around the discrete-space 
approximation applied in forest patch (or gap) models such as 
JABOWA, FORET, LINKAGES, and FORCLIM (Botkin et al. 
1972, Shugart and West 1977, Pastor and Post 1985, Bugmann 
1996, 2001).  Whereas these models simulate a set of trees 
on forest patches that do not interact, the model presented 
here can be executed with spatial interactions among patches, 
including dispersal of propagules and competition for light.

A noteworthy feature of the FOREL landscape model is 
its ability to address the effects of climate change on forest 
structure and dynamics.  Forest models simulating dynamics 
of trees in patches have long been used to investigate climate-
vegetation relationships (Botkin et al. 1972, Solomon et al. 
1981, Gates 1993).  Although some assumptions in several 
of these modeling investigations have been questioned (e.g., 
Bonan and Sirois 1992, Pacala and Hurtt 1993, Schenk 1996, 
Loehle and LeBlanc 1996, Reynolds et al. 2001), the criti-
cisms do not diminish the utility of this approach in general.  

Progress can be made with careful attention to critical assump-
tions concerning climate and tree performance (e.g., Bug-
mann and Solomon 2000).  Patch models, revised and tested, 
continue to advance the field of forest-climate dynamics 
(Gates 1993, Bugmann 2001).  Spatially-explicit projections 
at landscape levels allow the effects of large-scale patterns and 
processes to be considered and will further advance the field.

Purpose and scope
This report is aimed at describing the new landscape 

model, demonstrating its capabilities, and providing informa-
tion on its use.  The model is described in relation to other 
models of forest patch dynamics.  Capabilities of the model 
are then demonstrated using three distinct examples.  The 
first example shows spatial change in simulated forests along 
complex climate gradients.  The second shows spatial disper-
sal interactions among patches on a simulated landscape.  The 
third shows simulated forest response to temporal climatic 
change on a heterogeneous landscape.  Information on use 
of the model is provided along with copies of the computer 
program, sample input files, and a sample output file.

A Spatial Landscape Model of Forest Patch Dynamics and 
Climate Change

By Richard T. Busing

Abstract
FOREL (a FOREst Landscape model) is an individual-based, multi-scale simulator of forest and climate dynamics.  Ratio-

nale and design of the model are presented in relation to other forest patch models. Information on implementation of the model 
is also provided.  Capabilities of the FOREL model are demonstrated for forest composition, structure and dynamics along cli-
matic gradients.  The model relies on a patch simulation approach that has been tested and developed by independent ecologists 
for more than three decades.  Improvements made over the last decade to the simulation of climate effects on trees are incorpo-
rated in the landscape model.  A single parameterization of the model is capable of simulating major shifts in forest composition 
and structure across broad climatic gradients.  It is responsive along moisture gradients and temperature gradients.  The land-
scape model is flexible and can be altered easily to test various assumptions about the effects of climate on trees, and the effects 
of spatial pattern on processes operating within and among forest stands.  The spatial structure of the model makes interaction 
of patches possible.  Interactions may include dispersal of propagules and competition for light.  The model is a useful tool for 
projecting temporal climate change effects on forested sites, landscapes and regions.  

Methods and model constructs
In the new landscape model, simulation of patch dynam-

ics and climate effects on tree establishment, growth and 
mortality is similar to that in the FORCLIM model (version 
2.9; Bugmann and Solomon 2000).  The FORCLIM model 
has several improvements over forest patch models that were 
applied in early investigations of forest-climate dynamics.  
These improvements are discussed in the following section, 
and the ability of FORCLIM to simulate fundamental forest 
structure and composition along climate gradients is dem-
onstrated.  The weaknesses of FORCLIM are noted as well.  
They are addressed in development of the multi-scale land-
scape model.  

The patch dynamics framework

Modeling forest patch dynamics
Contemporary disturbance ecology has adopted a patch 

dynamics paradigm (Pickett and White 1985, Wu and Loucks 
1995).  A body of vegetation science theory has developed 
around this paradigm (van der Maarel 1996, 2005).  Because 
vegetation disturbance is often spatially discrete, the forma-
tion and development of patches in vegetation, including 
forests, has been a focus of research.  Modeling research, both 
theoretical and applied, has often relied on the disturbance 
patch as the fundamental spatial unit of simulation (Shugart 
and Smith 1996).  For example, Shugart (1984) referred to 
ecological forest patch simulators as “gap models” because 
they used patches that approximated the size of canopy gaps 
created by tree-fall disturbance (0.03-0.1 ha in area).  Thus 
the models were thought to simulate the forest growth cycle 
(sensu Whitmore 1982) of gap dynamics (Shugart 1984).  In 
comparison with forest models of higher spatial resolution, 
individual-based models of patch dynamics perform quite well 
at the stand level (Deutschman et al. 1999, Busing and Mailly 
2004).  In brief, the discrete-space approximation applied in 
these individual-based models of patch dynamics is expedient 
and effective at stand and landscape levels.

The FORCLIM patch model

General background and unique features
The general constructs of individual-based forest patch 

(or gap) models developed prior to 1990 are discussed by 
Shugart (1984) and Botkin (1992).  The individual-based 
design was revolutionary for a purely ecological model of 
forest dynamics (Botkin et al. 1972).  Insight on long-term 
dynamics of forests was gained through use of these models 
(Bormann and Likens 1979, Shugart 1984, Botkin 1992).  
However, the approach had some weaknesses.   First, some of 

the simplifying assumptions were either unproven or only par-
tially correct (Prentice 1994, Bugmann et al.1996).  Second, 
some simulation results diverged from reality (Huston 1992, 
Vanclay and Skovsgaard 1997).

Bugmann (1994, 1996), realizing the utility of this 
individual-based modeling approach for the simulation of 
climate-forest dynamics, developed a simplified patch model, 
which he called FORCLIM.  This model addressed several 
of the shortcomings in earlier forest models.  New features of 
FORCLIM included: 1) a cohort structure wherein trees of a 
cohort compete symmetrically with one another for light, 2) a 
modified tree growth equation with revised respiration costs 
(Moore 1989), and 3) a revision of growth suppression for 
trees in suboptimal environments.  In this way competition and 
its consequences for tree growth and mortality were altered.  
Subsequent revisions made to considerations of climate effects 
on trees led to a model that performed well in several distinct 
temperate regions of the world (Bugmann and Solomon 2000).  
The most noteworthy alterations to climate functions were: 1) 
the use of an asymptotic degree-day-growth relationship that 
reduces growth only at the cold edge(s) of a species range, 2) 
the use of a modified soil moisture bucket model that consid-
ers monthly water supply, demand and balance (Bugmann and 
Cramer 1998), 3) the inclusion of species-specific chilling 
requirements for establishment, and 4) consideration of the 
implications of contrasting leaf phenology between evergreen 
and deciduous species for tree growth.

Strengths of the FORCLIM version 2.9 modeling approach
In several aspects, the performance of FORCLIM 2.9 

was an improvement over earlier patch (or gap) models.  The 
first aspect of improved performance pertained to stand-level 
thinning of trees through competition.  In some patch models, 
the process of thinning from several trees to one dominant tree 
per patch was unrealistically rapid.  In such cases, reasonable 
thinning curves could not be obtained (Huston 1992).  How-
ever, the cohort structure and symmetry of competition for 
light among cohort members in FORCLIM reduced the rate 
of thinning.  As a result, thinning curves with realistic slopes 
could be obtained (Fig. 1).

Another benefit of the cohort structure in FORCLIM was 
the ability to produce a range of mortality-regeneration effects.  
When a single codominant tree died in FORCLIM, it had little 
effect on regeneration, but if an entire cohort of codominants 
died at once, there could have been a large increase in regener-
ation on the patch.  Thus, despite a constant patch size, a range 
of canopy effects and regeneration responses could occur.  In 
this way, the model may have been able to generate the variety 
of regeneration responses found in real forests where sizes and 
other characteristics of patches are not uniform (Busing and 
Mailly 2004).

�    A Spatial Landscape Model of Forest Patch Dynamics and Climate Change



Figure 1.  Thinning curve for simulated Pseudotsuga stands 
using FORCLIM.  The slope is approximately -3/2 for much of the 
curve and is in general agreement with studies on developing 
plant populations.

FORCLIM was designed to capture the effects of tempo-
ral changes in climate on forest composition.  If it is accurate 
in this regard it should also be able to reproduce general pat-
terns of compositional changes along spatial climate gradients.  
It has been shown to follow major spatial trends in temperate 
deciduous forest regions (Bugmann and Solomon 1995).  A 
subsequent version of FORCLIM (version 2.9) was well-
suited to temperate coniferous and deciduous forest regions 
as well (Bugmann and Solomon 2000).  Its ability to simulate 
effects of drought on tree growth and mortality was a notewor-
thy improvement over earlier ecological forest models.  In the 
Pacific Northwest, where winter temperatures can be mild and 
summer precipitation is low, spatial transitions along climate 
gradients within a climatically diverse region were simulated 
with FORCLIM 2.9 (Bugmann and Solomon 2000).  These 
compositional changes were shown to agree with field survey 
data (Busing and Solomon 2005).  For example, spatial cli-
mate-based changes in dominant tree species associated with 
elevation and continentality were successfully simulated (Figs. 
2, 3, & 4).

Figure 2.  Vegetation map of western Oregon showing the 
locations of sites along the study transect at 44.13o N latitude.

Figure 3.  Graph showing elevations of sites along the study 
transect at 44.13o N latitude.

Figure 4.  Graphs showing forest species composition and 
basal area along the study transect at 44.13o N latitude from (A) 
actual data (Busing 2004), and (B) simulated data projected by the 
FORCLIM model (Busing and Solomon 2005).

Weaknesses of the FORCLIM version 2.9 modeling 
approach

FORCLIM 2.9 is capable of simulating major trends in 
forest composition and total basal area across climate gradi-
ents (Busing and Solomon 2004, 2005, 2006).  However, ver-
sion 2.9 does not simulate some aspects of stand composition 
and structure with high accuracy.  One of the most obvious 
shortcomings of the model is the appearance of certain minor 
tree species in simulations at levels exceeding those measured 

in field surveys (Fig. 4).  The lack of dispersal limitations and 
the generality of climate functions are likely responsible for 
allowing unexpected subordinate species into simulated for-
ests.   A second shortcoming is the low accuracy of simulated 
tree diameter distributions.  A sharp decrease in stem density 
with tree diameter is simulated, but the number of small stems 
is underestimated by the model (Fig. 5).

Figure 5.  Graph showing actual and simulated size-class 
structure in the Cascades Montane Highlands ecoregion 
characterized by Pseudotsuga-dominated forests.  The simulated 
data are from FORCLIM model projections (Busing and Solomon 
2005).

Furthermore, the FORCLIM model is not spatially 
explicit.  It assumes that horizontal locations of stems and 
patches are not critical.  All that is known is what simulation 
patch each stem belongs to.  Interactions among patches are 
not simulated.  This approach does not handle propagule dis-
persal and other extended patch interactions well.   Horizontal 
interactions that affect tree establishment, growth and mortal-
ity are ignored or greatly simplified (Mailly et al. 2000, Bug-
mann 2001, Keane et al. 2001, Price et al. 2001).  How such 
omissions and simplifications affect simulation results is only 
beginning to be studied (Busing and Mailly 2004).   Other 
forest landscape models with interacting patches, including 
ZELIG (Smith and Urban 1988) and FIRE-BGC (Keane et al. 
1996), have revealed the importance of spatial considerations.  
For example, using ZELIG, Urban et al. (1991) demonstrated 
the potential for spatial effects (e.g., gap size and edge area) 
on tree species composition.  Development of a landscape 
model with spatial structure, patch interactions, and the ability 
to handle temperature and moisture gradients is discussed 
below.

The landscape model framework

Patch, block and landscape units
The spatial landscape model is a lattice of cells (or 

patches), each with unique Cartesian coordinates.  For efficient 
simulation of spatial interactions the Cartesian plane (or tract) 
is divided into square blocks containing square cells (Fig. 6).  
A cell represents a single simulation patch comparable to that 
used in many forest gap or patch models, but in this case its 
spatial location is known. Each cell belongs to a block of cells 
(e.g., cell A, Fig. 6).  All neighboring trees that compete with 
trees in a focal cell occur within the same block or within an 
adjacent block.  Thus only the focal block and adjacent blocks              
must be searched when looking for neighbors.  If a focal cell 
lies in a block at the edge of the simulation tract, repeating 
boundaries, also referred to as torus or translation methods 
(Haefner et al. 1991), are applied (e.g., cell B, Fig. 6).  In this 
case, adjacent blocks are taken from opposite edges of the 
simulation tract.

Figure 6.  Diagram showing nested structure of the landscape 
simulation grid (or tract).  Two cells (or patches) are highlighted 
in solid red.  Block boundaries are indicated by dashed lines.  
When extended neighborhood effects are employed, cells 
centered within the shaded area determined by the neighborhood 
radius are included.  Block size is set so that all cells within 
the neighborhood lie within the block of the focal cell or in 
blocks adjacent to that block.  For example, the search for cells 
neighboring cell A is restricted to blocks nearby.  Cell B belongs 
to an edge block.  In this case, neighboring cells may lie in blocks 
on the opposite side of the simulation tract.  The figure is adapted 
from Busing (1991).

Several kinds of spatial interactions can be modeled 
among cells.  They can include shading during tree establish-
ment, shading during tree growth, and dispersal of propagules.  

Simulation tract size, shape and structure
The model is easily modified to simulate a variety of tract 

sizes.  Tract shape is square.  Rectangular tracts require careful 
modification of the program, including the addition of vari-
ables.  For this reason, rectangular tracts are not recommended 
for preliminary simulation exercises.  Simulation cells (plots, 
patches, or gaps) are also square.  They lie within a square 
block.  Tract dimensions are multiples of block width.

The recommended patch dimensions of the model are 
30 X 30 m.   Thus, typical simulation cell (or patch) size is 
900 m2.  By comparison, the patch size commonly used in 
FORCLIM exercises is 833 m2.  Block width is a multiple 
of cell width.  Furthermore, blocks must be wide enough so 
that adjacent blocks include all neighboring cells that may 
be involved in competition for light.  Tract width must be a 
multiple of block width.  Because the repeating boundary 
approach is used for tract edges, it is recommended that tract 
width be at least several block widths in length.

Capabilities of the FOREL Model

To test and show the model’s capabilities, three general 
types of simulations were performed: 1) simulations along 
a transect with complex climatic gradients, 2) a simulation 
demonstrating the spatial dispersal capabilities of the model, 
and 3) a simulation of the response to climate change on a 
hypothetical landscape.

Transect simulations 
To demonstrate the climate driven nature of the model, 

simulations were run for eight sites along a transect from the 
coast to the high steppe in western Oregon (Busing and Solo-
mon 2005).  The west to east transect was located at 44.13o N 
and 121 to 124o W.  It spans a diverse climate gradient with 
strong intra-annual changes in temperature and precipitation 
(Fig. 7).   The middle portion of the transect includes high-ele-
vation sites (>1500 m) in the Cascade Range (Fig. 3), which 
feature cold conditions.   A site file with identical conditions 
except for monthly temperature and precipitation values, aver-
aged over a period of years (1961-1990), was used for each 
site (Bugmann and Solomon 2000, Busing and Solomon 2004, 
2005, 2006).  At each site ten simulations of 13 ha landscapes 
were run to the mean age of forest stands near the site (Bus-
ing and Solomon 2004, 2005).  A single species parameter 
file was used for all sites (Appendix B).  This file contains 
parameters similar to those used in the FORCLIM simula-
tions presented herein (Busing and Solomon 2006); however, 
new parameters have been added and existing parameters of a 
few species were modified.  Dispersal limitations were not in 

effect during this exercise.  Simulated climate variables and 
forest basal area along the transect were examined.

  

Figure 7.  Graphs showing climate changes along the study 
transect at 44.13o N latitude.  Monthly means of (A) temperature 
and (B) precipitation are provided for January (solid lines) and 
July (dotted lines).

Dispersal simulations
To demonstrate the spatial dispersal capabilities of the 

model, a simulated forest landscape (100 ha) dominated by 
Pseudotsuga was cut at 200 years of age with a 4-ha tract of 
retained forest in a portion of the landscape.  Propagule dis-
persal for recolonization was set so that only propagules from 
adult trees on the landscape could repopulate the vacant areas.  
In this example, Pseudotsuga had a mean dispersal distance of 
about 20 m from adult trees, and a negative exponential prob-
ability distribution was used to generate dispersal distances 
from adult trees. The landscape simulations were continued 

for a century after the cutting to allow examination of repopu-
lation trends and patterns.

Climate change simulations
To demonstrate the spatial climate change capabilities of 

the model on a heterogeneous landscape, a simulated forest 
landscape (ca. 400 ha) with topographic and climatic varia-
tion was subjected to temporal climate change. This required 
an advanced version of the model with unique environmental 
values (e.g., temperature and precipitation) for each patch (or 
cell) (model version 1.1).  Forest development from 0 to 400 
years was simulated with the baseline climate of transect site 
12 (Bugmann and Solomon 2000) (44.13o N, 121.96o W) on 
the lower west slope of South Sister, a volcanic peak in the 
Oregon Cascade Range.  Then from year 401 to 450 a linear 
increase of 3oC in temperature and 20% in precipitation was 
simulated.  The simulation was continued until 1000 years at 
the elevated temperature and moisture.  

This simulation was performed on a hypothetical moun-
tain landscape with slopes ranging from 25 to 50% and 
elevation differences of up to 1000 m (Fig. 8).  Slope aspects 
ranged from south-facing to north-facing.  Whereas baseline 
climate (transect site 12) represented the lower west slope 
of South Sister, the relatively steep upper west slope of this 
peak was represented by adjustments to the baseline climate.  
Precipitation increased with elevation (5.5 cm/month/1000m); 
temperature decreased with elevation (-6.7oC/1000m) (data in 
Bugmann and Solomon 2000).  Dispersal limitations were not 
in effect during this climate change exercise.  Furthermore, 
exogenous disturbances were not included in these simula-
tions. 

Figure 8.  Graph showing the simulation landscape surface of 
a hypothetical mountain with variation in elevation and slope 
aspect.  The x-axis runs north to south, and the y-axis runs east 
to west. Elevation increases along the y-axis (25% slope) and 
varies along the x-axis.  Slope aspect changes gradually along the 
x-axis from extreme north-facing when x equals zero and 2000 m, 
to neutral slope aspect when x equals 500 m and 1500 m, and to 
extreme south-facing when x equals 1000 m.

Results

Transect simulations

The simulations for eight sites along a complex climate 
gradient showed the changes in environmental conditions, total 
forest basal area, and dominant tree species known to occur in 
western Oregon.  The transect from the mild, wet coastal for-
ests, over the cool, wet montane and subalpine forests of the 
Cascade Range, and descending into cool, dry interior lands 
at intermediate elevations crosses many vegetation zones, 
including steppe lands that are too dry to support forests 
(Fig. 2).  The western end of the transect is characterized by 
high simulated degree-day sums (Fig. 9a) and low simulated 
drought (Fig. 9b).  This is particularly true in simulated sums 
for evergreen species, which dominate these forests character-
ized by high basal area.  The middle portion of the transect 
(across the Cascade Range) has low degree-day sums (Fig. 9a) 
and low drought (Fig. 9b).  Subalpine tree species dominate 
the cold, wet, sites at high elevations.  The eastern end of the 
transect is characterized by moderate degree-day sums (Fig. 
9a) and increasing drought (Fig. 9b).  Drought-tolerant species 
(e.g., Pinus ponderosa) dominate the eastern sites, but eventu-
ally give way to non-forest vegetation at the easternmost end 
of the transect, which is comparatively dry.

Trends in climate-based growth factors generated by the 
model for sites along the transect reveal how the growth com-
ponents of the model restrict the distribution of species.  For 
example, means of soil moisture growth factors and degree-
day growth factors for Pseudotsuga vary greatly along the 
transect (Fig. 10).  Low soil moisture inhibits growth of Pseu-
dotsuga at the two easternmost sites along the transect.  Low 
degree-day sums inhibit growth of Pseudotsuga in the high 
Cascades (site at 121.75o W).  Thus, Pseudotsuga is limited 
by either soil moisture or temperature at certain sites along 
the transect.  When a multiplicative growth factor is used (see 
Appendix A), as in FORCLIM, both low soil moisture and low 
degree-day sums can work in combination to reduce growth. 
This is evident for Pseudotsuga at certain sites with dimin-
ished basal area of this species (e.g., site at 121.5o W). 

Figure 9.  Graphs showing simulated climate variables along the 
study transect at 44.13o N latitude using the FOREL model.  Means 
of (A) annual degree-day sums and (B) annual drought indices are 
provided for deciduous (d, dotted lines) and evergreen species (e, 
solid lines).

Simulated and actual forests along the transect have 
similar trends in total basal area and in forest dominance (Fig. 
11).  Total basal area is high from the coast to the west slope 
of the Cascade Range, but declines in the subalpine forests and 
eastward to the steppe.  Moving west to east, dominance shifts 
from Pseudotsuga-Tsuga to Tsuga mertensiana and to Pinus 
ponderosa in both simulated and actual forests. 

Figure 10.  Graph showing simulated climate response variables 
affecting the growth of Pseudotsuga along the study transect at 
44.13o N latitude using the FOREL model.  Means of annual degree-
day growth factors (solid line), and annual soil moisture growth 
factors (dotted line) are provided.  Growth factor values less than 
1.0 reduce diameter growth; lower values cause greater reduction 
in growth.

Figure 11.  Graph showing simulated species composition and 
basal area along the study transect at 44.13o N latitude using the 
FOREL model.  See Fig. 4 for comparison to other simulated and 
actual forest data for these transect sites.

Dispersal simulations

The simulated landscape recolonization with limited 
dispersal from a retained patch of forest showed a potential 
for slow repopulation of the forested landscape (Fig. 12).  
The use of repeating boundaries resulted in recolonization of 
several areas of the simulated landscape.  However, portions 
of the landscape are not well populated, even after a century 
of recovery.  Thus, the model is capable of handling situations 
where dispersal of propagules limits recolonization and creates 
spatial patterns on the landscape.

Climate change simulations

Forest dynamics on a hypothetical mountain landscape 
representing the upper montane and subalpine areas of the 
western Cascades of Oregon exhibited a lagged response to 
climatic warming with increasing moisture.  On the mountain 
landscape (Fig. 8), the most striking response to the change in 
climate was the upslope movement of tree species (Fig. 13).  
Even though dispersal limitations were not simulated during 
this example, certain species did not advance rapidly upslope.  
Nearly three centuries after the episode of climatic warming 
(from 400-450 yr), Pseudotsuga had only begun to advance 
upslope into suitable sites (Fig. 13c).  Advancement of this 
species was still underway several centuries later (Fig. 13d).

Certain species retreated upslope within the simulation 
landscape.  For example, Abies lasiocarpa was widely distrib-
uted across the landscape prior to climatic change (Fig. 13a).  
It slowly became restricted to the highest elevations.  This 
migration process continued several centuries after climate 
change ceased. 

Model program structure and use

The new model consists of a FORTRAN program that 
reads two separate input files.  Site data are input from one 
file and species parameters are input from another file.  The 
FORTRAN file with comments and instructions is presented in 
Appendix A.  Example input files are presented in Appendix 
B.  

The FORTRAN program consists of a main loop that 
calls several subroutines, which, in turn, may call certain sub-
routines and functions.  Some of the subroutines are modifica-
tions of routines coded by earlier investigators for FORET, 
LINKAGES or FORCLIM.  The program was developed with 
Visual Fortran and it uses the intrinsic function RAN for pseu-
dorandom number generation.

Simulation parameters are set in the FORTRAN code, 
in the input files, and at console prompts during execution.  
Constants related to the simulated landscape tract dimensions 
and model functions are set in the FORTRAN program prior 
to compilation (see Appendix A).  Input file parameters are set 
in formatted ASCII text files prior to program execution.  The 
number of sites, model runs, simulation length, output interval, 

Figure 12.   Graphs showing landscape dynamics after cutting of a 200-year-old Pseudotsuga-dominated forest simulated by FOREL. 
A 4-ha patch of forest is retained on the 1-km2 tract.  Recolonization is initiated only by propagules dispersed from adult trees in the 
retention patch.  Landscape pattern is shown at (A) 20 years after cutting, (B) 50 years after cutting, and (C) 100 years after cutting.  
Horizontal units are in 30 m blocks.  The vertical axis is overstory height (m), which is exaggerated in relation to horizontal dimensions.

and output file name are entered from the console during 
execution of the program.  A single output file is generated 
during execution of the version documented herein.  

The program is designed to be flexible, allowing model 
sensitivity to various assumptions regarding propagule dis-
persal, tree establishment, spatial competition, tree diameter 
growth, tree height growth, and tree mortality to be tested.   
Options specified in the FORTRAN code and noted in com-
ments (Appendix A) include tract area and subdivisions, 
propagule dispersal limitations, delayed establishment, estab-
lishment rates, realized growth, competition for light extended 
beyond cell boundaries, disturbance, alternative height curves, 
mortality probability as a function of stress type, and mortality 
probability as a function of maximum age of a species.  The 
distance threshold for shading interactions can be specified (in 
subroutine CANOPY).  Mean dispersal distance is species-
specific.  It is a parameter that is input from the species file.  
Most simulation options also involve species-specific param-
eters from the species input file (Appendix B). 

Projections featuring temporal climate dynamics can be 
simulated as dictated by site file inputs.  The site file requires 
monthly means and standard deviations for present and future 
climates and the temporal break points (in simulation years) 
for climate changes (Appendix B).  Linear interpolation 
between pairs of break points is performed.  More than two 
break points can be input to accommodate complex climate 
dynamics.

The major functions of each subroutine are noted in the 
FORTRAN program comments (Appendix A).  Subroutines 
DISPERSE, CANOPY, DISTURB and OUTPUT are entirely 
new.  Functions of other subroutines resemble those docu-
mented elsewhere (Botkin et al. 1972, Shugart 1984, Pastor 
and Post 1985, Bugmann 1996, Bugmann and Solomon 2000).  
Those involving tree calculations are greatly restructured pri-
marily because of the spatial information required during sim-
ulations (Appendix A).  When enabled, subroutine DISPERSE 
restricts dispersal of propagules of each species to a select set 
of cells.  The set of cells is determined each simulation year 
from: 1) general propagule dispersal probabilities of the spe-
cies (in space and time), 2) local dispersal from adult trees of 
the species, or 3) both of these criteria.  Subroutine CANOPY 
evaluates canopy structure each simulation year.  Tree heights 
are estimated, and vertical shading leaf profiles are determined 
for each cell.  Two leaf profiles are generated: 1) a within-cell 
estimate, and 2) a neighborhood estimate that includes cells 
surrounding each focal cell.  The neighborhood estimate of 
shading leaf area can be used in subroutines ESTABLISH or 
GROW if desired.  Subroutine DISTURB simulates distur-
bance and assigns alternative death probabilities to affected 
trees.  In its simplest form it randomly selects cells each year 
and kills all trees in each selected cell.  An annual disturbance 
frequency greater than 0 and less than 1 must be set in this 
routine (Appendix A).  Subroutine OUTPUT writes live tree 
data to a text file (Appendix B).  Variables describing trees and 
their environment are output at intervals specified by the user 
during execution (see above).  One line is written per cohort 

of live trees in a given simulation year.  The fixed format file 
can be readily input into other software for analysis.  If an 
embedded approach to edge correction (Haefner et al. 1991) 
is desired, in addition to the correction provided, this can be 
accomplished simply by deleting edge cells from analyses of 
the output file after program execution.

Summary
An individual-based landscape model of forest and 

climate dynamics is presented.  The landscape model relies on 
a patch simulation approach that has been developed, tested 
and revised by various ecologists for more than three decades 
(Bugmann 2001).  Improvements made over the last decade 
to the simulation of climate effects on trees are incorporated 
in the landscape model. Capabilities of the model are demon-
strated for forest composition, structure and dynamics along 
diverse environmental gradients.  A single parameterization 
of the model is capable of simulating major shifts in forest 
composition and structure across broad climatic gradients.  It 
is sensitive to gradients of moisture and gradients of tempera-
ture.   Thus, it is likely to a useful tool for projecting temporal 
climate change effects on forested regions.  

The model is designed to be flexible.  It can easily be 
altered to test various assumptions about the effects of climate 
on trees, and the effects of spatial pattern on processes.  The 
spatial structure of the model makes interaction of proximate 
patches possible.  Such interactions may include dispersal of 
propagules and attenuation of light.  Options for various simu-
lation assumptions are discussed above and are clearly noted 
in the program listing (Appendix A).

Evaluation of various simulation assumptions and their 
effects on modeled projections requires further work.  Numer-
ous lines of investigation, ranging from vegetation composi-
tion to habitat and ecosystem dynamics, can be pursued.  Two 
areas requiring attention are propagule dispersal and the abun-
dance of stems of various sizes (Busing and Solomon 2005).  
The effects of various dispersal options require assessment and 
calibration to actual forest landscapes.  Improvement of size-
class structure of trees may be possible through examination 
and revision of assumptions and parameters concerning tree 
establishment, growth and mortality.  Work in these areas may 
broaden the realism and utility of the model.  

A basic framework for multi-scale simulation of forest 
dynamics is in place.  It simulates forest dynamics primar-
ily through effects of spatial competition and climate on 
individual trees.   The model is suitable for the study of local 
dynamics as well as pattern and process effects in forested 
landscapes.  

Summary    11

Appendix A Program listing

The entire FORTRAN program file is provided on the following pages.  It has been executed successfully in a Visual For-
tran Studio (2000 edition) for personal computers.  The active FORTRAN code is in black with intrinsic functions highlighted in 
blue, and line numbers in red.  The comments, in green, provide information on structure and operation of the program. 

C..... 
C.....A SPATIAL LANDSCAPE MODEL OF FOREST PATCH DYNAMICS AND CLIMATE 
C..... 
C.....R.T. BUSING 2005 
C..... 
C.....THE FOREL MODEL INCORPORATES HORIZONTAL SPATIAL CONSIDERATIONS 
C.....AMONG PATCHES IN A LATTICE MODEL OF GAP AND LANDSCAPE DYNAMICS. 
C.....PATCH (OR GAP) DYNAMICS ARE SIMULATED VIA THE DISCRETE-SPACE 
C.....APPROXIMATION APPLIED IN GAP MODELS SUCH AS JABOWA AND FORET, 
C.....BUT THE SIMULATION TRACT IS LARGER THAN THAT OF TRADITIONAL 
C.....GAP MODELS. IN THE CURRENT MODEL EACH PATCH IS SIMULATED WITHIN  
C.....THE LANDSCAPE MOSAIC AND CAN BE INFLUENCED BY PATCHES NEARBY.   
C.....REGENERATION IS REGULATED BY EXOGENOUS AND ENDOGENOUS SOURCES.   
C.....DISPERSAL LIMITATION CAN BE EMPLOYED.  TREE GROWTH AND CLIMATE 
C.....RELATIONSHIPS ARE SIMILAR TO THOSE OF THE FORCLIM MODEL.  THE 
C.....MODEL IS INDIVIDUAL BASED WITH EACH TREE STORED IN A COHORT  
C.....ARRAY. MORE THAN ONE COHORT CAN OCCUPY A GRID CELL (ONE CELL  
C.....IS A PATCH SIMULATION UNIT).  A SPECIES-SPECIFIC LAG TIME  
C.....BETWEEN ESTABLISHMENT AND RECRUITMENT (AT 1.27 CM DBH) CAN BE   
C.....EMPLOYED. THERE ARE NINE SHADE TOLERANCE CLASSES.  TO EXPEDITE   
C.....SEARCHES FOR NEIGHBORS, ARRAYS OF COHORTS ARE PARTITIONED INTO  
C.....BLOCKS REPRESENTING PORTIONS OF THE SIMULATION TRACT. 
C..... 
C.....VERSION 1.0 
C..... 
C.....USAGE: THIS PROGRAM WAS DEVELOPED AS A CONSOLE APPLICATION IN  
C.....VISUAL FORTRAN WITH THE LIBRARY FUNCTION RAN.  TWO INPUT FILES  
C.....ARE REQUIRED: A SPECIES FILE SPECIFYING THE SPECIES POOL AND  
C.....PARAMETERS FOR EACH TREE SPECIES, AND A SITE FILE WITH 
C.....ENVIRONMENTAL CONDITIONS, INCLUDING CLIMATE.  BASIC SIMULATION  
C.....PARAMETERS ARE INPUT FROM CONSOLE PROMPTS DURING EXECUTION.   
C.....CERTAIN PROGRAM FEATURES CAN BE READILY ENABLED OR DISABLED.   
C.....IMPORTANT OPTIONS IN THIS REGARD ARE NOTED IN THE PROGRAM  
C.....COMMENTS WITH THE WORD "OPTIONS".  
C..... 

PARAMETER (MT=350,MB=20,MS=20)
C..... 
C.....MT - THE MAXIMUM NUMBER OF COHORTS PER SQUARE BLOCK
C.....MB - THE MAXIMUM NUMBER OF BLOCKS ON A SIDE OF TRACT 
C.....MC - THE MAXIMUM NUMBER OF CELLS ON A SIDE OF TRACT 
C.....MS - THE MAXIMUM NUMBER OF SPECIES 
C..... 

COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB),
 
>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB),

>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB),

>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB),

>NOROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)

COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS), 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),
>DPRBL(MS),DDMIN(MS),WITN(MS),WITX(MS),
>IBRW(MS),KDECID(MS)
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
COMMON NSELCT(MS)
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
COMMON/SEED/USEED(15)
INTEGER USEED 
CHARACTER*25 FNAME 

C..... 

C.....COHORT ARRAY VARIABLES:  

C..... 
C.....NCOHS - NUMBER OF TREE COHORTS FOR EACH SPECIES 
C.....DBH - DIAMETER AT BREAST HEIGHT OF TREES IN COHORT 
C.....TLA - LEAF AREA OF EACH COHORT 
C.....TMASS - BIOMASS OF EACH COHORT 
C.....HTCM - HEIGHT OF EACH COHORT 
C.....LOCX & Y - LATTICE LOCATION OR CELL OF EACH COHORT 
C.....ISPEC - SPECIES OF EACH COHORT 
C.....NOGRO - COHORTS THAT GROW VERY SLOWLY 
C.....NOGROL - COHORTS THAT ARE NOGRO BECAUSE OF LOW LIGHT 
C.....NOGROW - COHORTS THAT ARE NOGRO BECAUSE OF DROUGHT 
C.....NOGROT - COHORTS THAT ARE NOGRO BECAUSE OF COLD 
C.....IAGE - AGE OF EACH COHORT 
C.....ADP - ANNUAL DEATH PROBABILITY FOR TREES IN EACH COHORT 
C.....ICOHSIZ - NUMBER OF TREES IN COHORT 
C..... 
C.....RANDOM NUMBER SEEDS AND USAGE: 
C..... 
C.....USEED(1) - KILL: TEST MORTALITY FOR EACH TREE 
C.....USEED(2) - KILL: TEST FOR STRESSED (NOGRO) TREE DEATH 
C.....USEED(3) - ESTABLISH: BROWSE FILTER FACTOR AT ESTABLISHMENT 
C.....USEED(4) - ESTABLISH: PROBABILISTIC ESTABLISHMENT FILTER  
C.....USEED(5) - ESTABLISH: COHORT SIZE FACTOR AT ESTABLISHMENT 
C.....USEED(6) - DISPERSE: PROBABILISTIC EXOGENOUS DISPERSAL 
C.....USEED(7) - DISPERSE: DISPERSAL DISTANCE FACTOR
C.....USEED(8) - DISPERSE: DISPERSAL DIRECTION 
C.....USEED(9 TO 12) - CALL GGNORD: TEMP & PRECIP VARIATION
C.....USEED (13) - DISTURB: PROBABILISTIC DISTURBANCE 
C..... 
C.....INITIAL SEEDS FOR RANDOM NUMBER FUNCTION RAN 
C.....RAN IS A VISUAL FORTRAN LIBRARY FUNCTION THAT 
C.....GENERATES NUMBERS UNIFORM OVER RANGE: GE 0 LT 1 
C.....LARGE ODD SEEDS ARE RECOMMENDED 
C.....FOLLOWS ALGORITHM OF PARK & MILLER 1988 
C..... 

USEED(1) = 18641

USEED(2) = 34463

USEED(3) = 46647

USEED(4) = 90733

USEED(5) = 45199

USEED(6) = 12567

USEED(7) = 21373

USEED(8) = 34021

USEED(9) = 31563

USEED(10) = 54889

USEED(11) = 76095

USEED(12) = 43531

USEED(13) = 92355

USEED(14) = 58631

USEED(15) = 83983


C..... 
CALL READSPEC 

C..... 

C.....SET SIMULATION PARAMETERS
 
C..... 

C.....NSITE - NUMBER OF SITES, IF >1, GIVE MULTIPLE SITES IN INPUT FILE 

C.....NRUN - NUMBER OF RUNS PER SITE  

C.....NYEAR - LENGTH OF RUN (YR)

C.....KPRNT - PRINT INTERVAL (YR)

C..... 




C:\FORELdoc\FOREL1.F Page 3 of 28 

file://C:\DOCUME~1\busingr\LOCALS~1\Temp\FOREL1_F.htm 2/13/2007 

A Spatial Landscape Model of Forest Patch Dynamics and Climate Change Appendix A  17

WRITE (*,14)
14 	 FORMAT(1X,'INPUT NUMBER OF SITES')
 

READ (*,16)NSITE

WRITE (*,15)
 

15 FORMAT(1X,'INPUT NUMBER OF RUNS')
 
READ (*,16)NRUN
 

16 FORMAT (I3)
 
WRITE (*,17)
 

17 FORMAT(1X,'INPUT NUMBER OF YEARS')
 
READ (*,18)NYEAR
 

18 FORMAT (I5)
 
WRITE (*,19)
 

19 	 FORMAT (1X,'INPUT PRINT INTERVAL IN YEARS')
 
READ (*,18)KPRNT

WRITE (*,20)
 

20 	 FORMAT (1X,'GIVE FILE NAME FOR OUTPUT')
 
READ (*,'(A)') FNAME
 
OPEN (6,FILE=FNAME,STATUS='NEW')

REWIND 6 


C..... 
C.....OPTIONS: LANDSCAPE VARIABLES TO BE SET BY USER: 
C..... 
C.....CELLAREA - SQUARE PATCH (OR CELL) AREA IN SQ M UNITS, CELL WIDTH 
C.....IS SQRT (CELLAREA) 
C.....KWIDTH - SQUARE BLOCK WIDTH IN SQRT(CELLAREA) UNITS, MUST BE 
C.....GREATER THAN OR EQUAL TO 1
C.....IWIDTH - SQUARE TRACT WIDTH IN SQRT(CELLAREA) UNITS, MUST BE A  
C.....MULTIPLE OF KWIDTH 
C.....NBLKS - THE NUMBER OF BLOCKS ON A SIDE OF THE SQUARE TRACT 
C.....TOTAR - TRACT AREA IN HA 
C..... 

CELLAREA = 900.
 
IWIDTH = 12
 
KWIDTH = 3
 
NBLKS = IWIDTH/KWIDTH

TOTAR = ((SQRT(CELLAREA)*IWIDTH)**2)/10000. 

DO 200 IS = 1,NSITE
 

C..... 

C.....LOOP THROUGH SITES 

C.....READ SITE ENVIRONMENT INPUTS VIA SUBR READSITE
 
C..... 


CALL READSITE(IS,NYEAR)
 
DO 200 IRUN = 1,NRUN
 

C..... 

C.....LOOP THROUGH RUNS (MULTIPLE SIMULATION TRACTS FOR SITE)

C.....CLEAR ARRAYS, AND PROCEED 

C..... 


CALL RUNIN
 
DO 200 KYR = 1,NYEAR
 

C..... 

C.....LOOP THROUGH SIMULATION YEARS AT ONE YEAR TIME STEP
 
C.....ESTIMATE CLIMATE FACTORS THEN CYCLE THROUGH KILL, DISPERSE,  

C.....CANOPY, SEEDLIN & GROW SUBROUTINES ANNUALLY.  GENERATE 

C.....TEMPERATURE & PRECIPITATION EACH YEAR, WITH RANDOM VARIATION  

C.....INDEPENDENT AMONG MONTHS, YRS & RUNS.  

C..... 

C.....OPTIONS: FOR INDEPENDENT CELL MODE SKIP DISPERSE ENTIRELY, SKIP  

C.....AVAILABILITY REQUIREMENT LINE IN ESTABLISH, AND USE LOCAL SHADING 

C.....LEAF SUMS (SLAS) IN ESTABLISH & GROW. 

C..... 

WRITE(*,2002)IS,IRUN,KYR
2002	 FORMAT(1X,'SITE ',I4,2X,'RUN ',I4,2X,'YEAR ',I5)
 

CALL TEMPE
 
CALL WATER 

CALL KILL
 
CALL CANOPY
 
CALL DISPERSE
 
CALL ESTABLISH
 
CALL GROW(IRUN)
 
IF(((KYR/KPRNT)*KPRNT).EQ.KYR) CALL OUTPUT(IS,IRUN)
 

200 CONTINUE
 
STOP
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE READSPEC 
C..... 

C.....SUBROUTINE READSPEC INPUT SPECIES PARAMETERS FROM A FILE
 
C.....ONCE AT THE BEGINNING OF EACH SIMULATION EXECUTION.
 
C..... 


PARAMETER (MT=350,MS=20)
COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS), 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),
>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),
>WITX(MS),IBRW(MS),KDECID(MS)
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
COMMON NSELCT(MS)

C..... 
C.....THE SPECIES INPUT FILE IS SP*.DAT 
C.....NSPEC - NUMBER OF SPECIES TO BE INCLUDED 
C.....NSELCT - ARRAY OF SPECIES NUMBERS FOR THOSE SPECIES 
C..... USED IN THE SIMULATION 
C..... 

OPEN(UNIT=3,FILE='SPNEW6.DAT')

READ(3,9000) NSPEC,(NSELCT(I),I=1,NSPEC)
 

C..... 
C.....INPUT INDIVIDUAL SPECIES PARAMETERS AS SPECIES ARRAY CONSTANTS 
C.....AAA - SPECIES FOUR LETTER CODE NAME 
C.....DMX - SPECIES MAX DBH CM 
C.....HMX - SPECIES MAX HEIGHT M 
C.....AGEMX - MAX AGE OF SPECIES YR 
C.....ITOL - SPECIES LIGHT TOLERANCE CLASS 
C.....NTOL - SPECIES NITROGEN TOLERANCE CLASS 
C.....DRTOL - SPECIES DOUGHT TOLERANCE 
C.....G - SPECIES GROWTH RATE SCALING CONSTANT 
C.....LTIME - AGE OF SPECIES AT RECRUITMENT DIAMETER 
C.....DISP - PROBABILITY OF EXOGENOUS SEED DISPERSAL FOR SPECIES 
C.....DISD - MEAN SEED DISPERSAL DISTANCE FOR SPECIES (CELLWIDTH UNITS) 
C.....DREPR - SPECIES MINIMUM DBH FOR HIGH SEED PRODUCTION  
C.....YLI - LIGHT DEMAND OF SEEDLING SPECIES 
C.....ESTMX - SPECIES MAX SEEDLING ANNUAL INFLUX 
C.....FC0 - SPECIES FOLIAGE AREA CONSTANT (FORCLIM VERS. 2.9) 
C.....FC1 - SPECIES FOLIAGE AREA CONSTANT 1 
C.....FC2 - SPECIES FOLIAGE AREA CONSTANT 2 
C.....HC0 - SPECIES ASYMPTOTIC HEIGHT CONSTANT 0  
C.....HC1 - SPECIES ASYMPTOTIC HEIGHT CONSTANT 1 
C.....HC2 - SPECIES ASYMPTOTIC HEIGHT CONSTANT 2   
C.....DPRB - SPECIES DEATH PROB UNDER STRESS  

C.....DPRBL - SPECIES DEATH PROB UNDER LIGHT STRESS 
C.....DDMIN - SPECIES MINIMUM DEGREE-DAY REQUIREMENT
C.....WITN - SPECIES WINTER TEMPERATURE MIN FOR ESTABLISHMENT 
C.....WITX - SPECIES WINTER CHILLING REQUIREMENT FOR ESTABLISHMENT  
C.....IBRW - SPECIES IMPACT OF BROWSING ON ESTABLISHMENT 
C.....KDECID - SPECIES TYPE FLAG, DECIDUOUS EQUALS 1
C.....NUM - INDIVIDUAL SPECIES NUMBER 
C..... 

J = 1
 
DO 10 K=1,100


READ(3,9001) AAA(J),DMX(J),HMX(J),AGEMX(J),ITOL(J),NTOL(J), 
> DRTOL(J),G(J),LTIME(J),DISP(J),DISD(J),DREPR(J),YLI(J), 
> IBRW(J),ESTMX(J),FC0(J),FC1(J),FC2(J),HC0(J),HC1(J),HC2(J), 
> DPRBL(J),DDMIN(J),WITN(J),WITX(J),KDECID(J),NUM

IF (NSELCT(J).NE.NUM) GO TO 10 
WRITE(6,9002) AAA(J),DMX(J),HMX(J),AGEMX(J),ITOL(J),NTOL(J), 

> DRTOL(J),G(J),LTIME(J),DISP(J),DISD(J),DREPR(J),YLI(J), 
> IBRW(J),ESTMX(J),FC0(J),FC1(J),FC2(J),HC0(J),HC1(J),HC2(J), 
> DPRBL(J),DDMIN(J),WITN(J),WITX(J),KDECID(J),NUM

IF (NUM.EQ.NSELCT(NSPEC)) GO TO 20
 
J = J+1


 10 CONTINUE

 20 CONTINUE
 

RETURN

 9000 FORMAT(40I2)
 
9001 FORMAT(1A4,2F4.0,F5.0,2I2,F5.2,F4.0,I3,F5.2,F4.1,F3.0,F5.2,I2,/
 

> F5.3,2F6.2,F5.2,F5.1,F6.3,F5.2,F5.3,F5.0,2F4.0,I2,I3) 
9002 FORMAT(' ',1A4,2F5.0,F6.0,2I3,F6.2,F5.0,I3,F5.2,2F5.1,F5.2,I2,(/), 

> F5.3,2F6.2,F5.2,F5.1,3F6.3,F5.0,2F4.0,I2,I3)
END 

C..... 
C..... 
C..... 

SUBROUTINE READSITE(IS,NYEAR)
C..... 
C.....SUBROUTINE READSITE READS SITE PARAMETERS, LATITUDE, SOIL  
C.....BUCKET SIZE, MAXIMUM EVAPOTRANSPIRATION, SLOPE ASPECT FACTOR,  
C.....AS WELL AS MONTHLY TEMPERATURE, PRECIPITATION & CORRESPONDING  
C.....STANDARD DEVIATIONS FROM A FILE.  VALUES FOR INTERANNUAL 
C.....INTERPOLATION OF CLIMATE CAN BE SPECIFIED.  READSITE IS CALLED 
C.....ONCE PER SITE TO PROVIDE ENVIRONMENTAL CONDITIONS FOR EACH  
C.....LANDSCAPE TRACT. 
C..... 

COMMON/INTERP/IPOLAT,X(10)
COMMON/WEATH/T(12),VT(12),RT(13),R(12),VR(12),BS,CW,SLASP,PLAT, 

>AET,DDE,DDD,TCOLD,DRE,DRD 
COMMON/LINEAR/TSAV(45,12),VTSAV(45,12),RSAV(45,12),VRSAV(45,12) 

C..... 

C.....PROVIDE NAME OF SITE FILE TO OPEN
 
C..... 


OPEN(UNIT=5,FILE='WORTR13.DAT',ACCESS='SEQUENTIAL',STATUS='OLD') 
C..... 

C.....IPOLAT HOLDS THE NUMBER OF BREAK POINTS IN LINEAR INTERPOLATIONS 

C.....(EQUALS NUMBER OF ENTRIES IN ARRAY X, & THE NUMBER OF LINES OF  

C.....INPUT PER CLIMATE VARIABLE TYPE)

C..... 


READ(5,1010) IPOLAT
 
1010 FORMAT(7X,I2)
 

C..... 

C.....CLIMATE CHANGE SIMULATIONS REQUIRE CHANGING CLIMATE MEANS OR STD  

C.....DEVS. ARRAY X CONTAINS YEARS CORRESPONDING TO NEW CLIMATE VALUES.  

C.....LINEAR INTERPOLATIONS ARE MADE BETWEEN VALUES. THE FIRST ENTRY 

C.....MUST BE EQUAL TO ZERO AND THE LAST EQUAL TO NYEAR.

C..... 


READ(5,1015) (X(I),I=1,10)
 
1015 FORMAT(10F7.0)
 

IF(X(IPOLAT).LT.NYEAR) STOP 'READSITE NYEAR>LAST X ARRAY VALUE'
 
C..... 

C.....READ LATITUDE, LONGITUDE, BUCKET SIZE, MAX EVAPOTRANS RATE, 

C.....AND SLOPE ASPECT FACTOR CF FORCLIM2.9 

C..... 


READ(5,1020) PLAT,BS,CW,SLASP
 
1020 FORMAT(5X,F5.2,4X,F5.1,4X,F5.1,7X,F5.2)
 

C..... 

C.....READ MONTHLY TEMPERATURE, STD DEV, PRECIPITATION, STD DEV 

C..... 


READ(5,1035) ((TSAV(J,K),K=1,12),J=1,IPOLAT)

READ(5,1035) ((VTSAV(J,K),K=1,12),J=1,IPOLAT)

READ(5,1035) ((RSAV(J,K),K=1,12),J=1,IPOLAT)

READ(5,1035) ((VRSAV(J,K),K=1,12),J=1,IPOLAT)
 

1035 FORMAT(8X,12F6.0)
C..... 

C.....WRITE CLIMATE DATA  

C..... 


WRITE(*,1025) IS,PLAT,BS,CW,SLASP 
1025 FORMAT('SITE=',I4,' LATITUDE=',F5.1,
 

1' BUCKET SIZE=',F5.1,
 
2' MAX EVAPOTR=',F5.1,
 
3' SLOPE ASPEC=',F5.1)

WRITE(*,7036)

7036 FORMAT(' '/12X,'J',5X,'F',5X,'M',5X,'A',5X,'M',5X,'J',5X,'J',5X, 
1'A',5X,'S',5X,'O',5X,'N',5X,'D')
WRITE(*,7037) (TSAV(1,K),K=1,12) 

7037 FORMAT('TEMP (C)',12F6.1)
WRITE(*,7038) (VTSAV(1,K),K=1,12) 

7038 FORMAT('STND DEV',12F6.1)
WRITE(*,7039) (RSAV(1,K),K=1,12) 

7039 FORMAT('PPT (CM)',12F6.1)
WRITE(*,7040) (VRSAV(1,K),K=1,12) 

7040 FORMAT('STND DEV',12F6.1/)
 
RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE RUNIN 
C..... 

C.....SUBROUTINE RUNIN INITIALIZES VARIABLES.  IT IS CALLED AT
 
C.....THE BEGINNING OF EACH RUN OF A LANDSCAPE TRACT.
 
C..... 


PARAMETER (MT=350,MB=20,MS=20)
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB),
 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB),

>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB),

>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB),

>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)

COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
 
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
 

C..... 

C.....INITIALIZE VARIABLES TO START SIMULATION ON BARE TRACT

C..... 
NFULL = 0
 
DO 10 I=1,NSPEC


NCOHS(I) = 0
 
10 CONTINUE
 

DO 20 JX=1,NBLKS

DO 20 JY=1,NBLKS


NTOT(JX,JY)=0

DO 20 I=1,MT

DBH(I,JX,JY) = 0.

HTCM(I,JX,JY) = 0.

TMASS(I,JX,JY) = 0.

TLA(I,JX,JY) = 0.

NOGRO(I,JX,JY) = 0

NOGROL(I,JX,JY) = 0

NOGROW(I,JX,JY) = 0

NOGROT(I,JX,JY) = 0

ISPEC(I,JX,JY) = 0

ICOHSIZ(I,JX,JY) = 0

ADP(I,JX,JY) =0.
 

C..... 

C.....IAGE IS -1 FOR DEAD COHORTS
 
C..... 


IAGE(I,JX,JY) = -1 
C..... 

C.....LOC IS INITIALIZED OUT OF RANGE
 
C..... 


LOCX(I,JX,JY) = 99999

LOCY(I,JX,JY) = 99999
 

20 CONTINUE
 
RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE KILL 
C..... 

C.....SUBROUTINE KILL EVALUATES EACH TREE AND REMOVES A FEW PURELY 

C.....BY CHANCE. THOSE HAVING PERIODS OF SLOW GROWTH LASTING MORE 

C.....THAN ONE YEAR ARE SUBJECT TO HIGHER RISK OF DEATH.  SPECIES  

C.....RISK OF DEATH FROM LIGHT STRESS CAN BE SPECIFIED IN DPRBL. 

C.....THIS SUBROUTINE IS CALLED BY MAIN ONCE EACH YEAR.  IT IS A 

C.....MODIFICATION OF THE KILL SUBROUTINE IN THE FORET MODEL.
 
C..... 


PARAMETER (MT=350,MB=20,MS=20)
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB), 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB), 
>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB), 
>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB), 
>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)
COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS), 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),
>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),
>WITX(MS),IBRW(MS),KDECID(MS)
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS 
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR 
COMMON/SEED/USEED(15)
INTEGER USEED 
IF(NFULL.EQ.0) RETURN 

C..... 
C.....CALL SUBR DISTURB FOR ANNUAL DEATH RISKS SUPERCEDING THE 
C.....DEFAULT RISKS, WHICH ARE DERIVED FROM SPECIES MAXIMUM AGE 
C.....OPTIONS: FOR MORTALITY AT LOW LEVELS DERIVED FROM 
C.....MAX AGE OR STRESS SKIP THIS CALL   
C..... 
C CALL DISTURB 
C..... 
C.....LOOP THROUGH BLOCKS 
C..... 

DO 30 KX=1,NBLKS

DO 30 KY=1,NBLKS
 

C..... 

C.....LOOP THROUGH COHORTS IN BLOCK
 
C..... 


DO 30 K=1,NTOT(KX,KY) 

I=ISPEC(K,KX,KY)

ITRE=ICOHSIZ(K,KX,KY)
 

C..... 

C.....OPTIONS: NEXT LINE TURNS OFF MORTALITY IN SEEDLING POOL
 
C..... 

C IF(DBH(K,KX,KY).EQ.0.) GO TO 30
 
C..... 

C.....LOOP THROUGH TREES IN COHORT
 
C..... 


DO 25 ITR=1,ITRE
C..... 

C.....KILL TREES RANDOMLY 

C.....UNDER DEFAULT BACKGROUND LEVELS FEW REACH MAX AGE
 
C..... 


IF (RAN(USEED(1)).LE.(ADP(K,KX,KY))) GO TO 15 
C..... 

C.....SLOW GROWTH FOR TWO CONSECUTIVE YEARS INCREASES DEATH RISK  

C.....CF SOLOMON 1986
 
C..... 


IF (NOGRO(K,KX,KY).LT.2) GO TO 25 
C..... 

C.....ADDITIONAL DEATH RISK FOR NOGRO TREES IS SET BELOW
 
C..... 


DPROB=0.368 
C..... 

C.....IF STRESS IS FROM LOW LIGHT USE A SPECIES-SPECIFIC DEATH RISK
 
C..... 


IF (NOGROL(K,KX,KY).GT.0) DPROB=DPRBL(I) 
C..... 

C.....IF STRESS IS FROM COLD USE DEATH RISK SPECIFIED HERE
 
C..... 


IF (NOGROT(K,KX,KY).GT.0) DPROB=0.368 
C..... 

C.....IF STRESS IS FROM DROUGHT USE DEATH RISK SPECIFIED HERE
 
C..... 


IF (NOGROW(K,KX,KY).GT.0) DPROB=0.368 

IF (RAN(USEED(2)).GT.DPROB) GO TO 25
 

C..... 

C.....IF KILLED DECREMENT NUMBER OF TREES IN COHORT 

C..... 


15 ICOHSIZ(K,KX,KY) = ICOHSIZ(K,KX,KY)-1 
C..... 

C.....TO NEXT LIVE TREE IN COHORT ITRE

C..... 
25 CONTINUE 

C..... 
IF (ICOHSIZ(K,KX,KY).GT.0) GO TO 30 

C.....IF EMPTY COHORT DECREMENT NUMBER OF COHORTS FOR SPECIES 
C..... 

NCOHS(I) = NCOHS(I)-1 
C..... 

C.....TEMPORARILY FLAG DEAD COHORTS
 
C..... 


IAGE(K,KX,KY) = -1
C..... 
C.....RESET ANNUAL DEATH PROB TO BACKGROUND LEVEL FOR SPECIES 
C.....THIS LEVEL IS INITIALIZED IN ESTABLISH 
C.....5.3 GIVES APPROXIMATELY 1/2% SURVIVAL TO MAX AGE
C.....4.6 GIVES APPROXIMATELY 1% SURVIVAL TO MAX AGE 
C.....4.0 GIVES APPROXIMATELY 2% SURVIVAL TO MAX AGE 
C.....FORET & FORCLIM DEFAULT VALUE 4.605 
C..... 
C.....TO NEXT LIVE COHORT K 
C..... 

30 ADP(K,KX,KY) = 4.6/AGEMX(I) 
C..... 
C.....LOOP THROUGH BLOCKS AND COHORTS BY BLOCK 
C.....REWRITE ARRAY VARIABLES TO ELIMINATE DEAD COHORTS 
C.....AND TALLY K LIVE COHORTS IN EACH BLOCK 
C..... 

DO 40 IX=1,NBLKS
DO 40 IY=1,NBLKS
K=0 
DO 35 I=1,NTOT(IX,IY)
IF (IAGE(I,IX,IY).EQ.-1) GO TO 35 
K = K+1 
DBH(K,IX,IY) = DBH(I,IX,IY)
IAGE(K,IX,IY) = IAGE(I,IX,IY)
NOGRO(K,IX,IY) = NOGRO(I,IX,IY)
NOGROL (K,IX,IY) = NOGROL(I,IX,IY)
NOGROW(K,IX,IY) = NOGROW(I,IX,IY)
NOGROT(K,IX,IY) = NOGROT(I,IX,IY)
ISPEC(K,IX,IY) = ISPEC(I,IX,IY)
LOCX(K,IX,IY) = LOCX(I,IX,IY)
LOCY(K,IX,IY) = LOCY(I,IX,IY)
TLA(K,IX,IY) = TLA(I,IX,IY)
HTCM(K,IX,IY) = HTCM(I,IX,IY)
TMASS(K,IX,IY) = TMASS(I,IX,IY)
ICOHSIZ(K,IX,IY) = ICOHSIZ(I,IX,IY)
ADP(K,IX,IY) = ADP(I,IX,IY) 

35 CONTINUE 
NTOT(IX,IY)=K

40 CONTINUE 
NFULL=0 
DO 50 IS=1,NSPEC

50 NFULL=NFULL+NCOHS(IS)
RETURN 
END 

C..... 
C..... 
C..... 

SUBROUTINE ESTABLISH 
C..... 

C.....SUBROUTINE ESTABLISH REGULATES SPECIES ESTABLISHMENT IN EACH CELL 
C.....VIA A SERIES OF FILTERS.  IT IS CALLED BY MAIN ONCE EACH YEAR. 
C..... 

PARAMETER (MT=350,MB=20,MS=20,MC=33) 
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB),
 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB),

>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB),

>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB),

>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)

COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS), 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),
>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),
>WITX(MS),IBRW(MS),KDECID(MS)
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
COMMON/WEATH/T(12),VT(12),RT(13),R(12),VR(12),BS,CW,SLASP,PLAT, 

>AET,DDE,DDD,TCOLD,DRE,DRD
COMMON/PATCH/IFLAG(MS,MC,MC),SLA(1300,MC,MC),SLAX(1300,MC,MC)
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
COMMON/SEED/USEED(15)
INTEGER USEED 

C..... 

C.....INPUTS SAPLINGS FOR CELLS FLAGGED AS HAVING PROPAGULES 

C..... 

C.....INDIVIDUALS ENTER THE SEEDLING POOL AT 0 CM DBH 

C.....RECRUITMENT OCCURS AFTER A SPECIES-SPECIFIC TIME LAG (LTIME)  

C.....ONLY THEN IS DBH INCREASED TO 1.27 CM IN SUBROUTINE GROW, 

C.....CF BUSING 1995 

C..... 


SIZE = 0.0
 
PHI=1.0
 

C..... 

C.....LOOP THROUGH CELLS EVALUATING CONDITIONS AND ELIGIBILITY FOR 

C.....ESTABLISHMENT, CELLS ARE SPECIFIED BY COORDINATES LX,LY 

C..... 


DO 85 LX=1,IWIDTH
 
DO 85 LY=1,IWIDTH
 

C..... 

C.....DETERMINE WHICH BLOCK IX,IY EACH CELL IS IN  

C.....EACH COHORT IS STORED IN A BLOCK
 
C..... 


IX=((LX-1)/KWIDTH)+1

IY=((LY-1)/KWIDTH)+1

IF(IX.LT.1.OR.IY.LT.1) STOP 'LOW BLOCK ERROR IN ESTABLISH'
 
IF(IX.GT.NBLKS.OR.IY.GT.NBLKS) 


> STOP 'HIGH BLOCK ERROR IN ESTABLISH' 
C..... 

C.....DETERMINE SHADING CANOPY ABOVE FLOOR OF CELL 

C.....AVAILABLE LIGHT ATTENUATION CONSTANT IS 0.25, BUGMANN 1994 

C..... 


SLAS=0.0 
C..... 

C.....SLAS IS THE LEAF AREA DIRECTLY ABOVE THE CELL 

C.....SLAXS IS THE AVERAGE LEAF AREA PER CELL WITHIN KRADIUS OF CELL 

C..... 


SLAXS=0.0
 
DO 22 JH=1,1300

SLAS=SLAS+SLA(JH,LX,LY)
 

22	 SLAXS=SLAXS+SLAX(JH,LX,LY)

SLAS=AMAX1(SLAS,0.0) 

AL = PHI*EXP(-0.25*(SLAS/CELLAREA))
 
SLAXS=AMAX1(SLAXS,0.0)
 
ALX =PHI*EXP(-0.25*(SLAXS/CELLAREA))
 

DO 75 ISP=1,NSPEC
C..... 

C.....ARE PROPAGULES AVAILABLE FOR THIS SPECIES IN THIS PATCH?
 
C.....OPTIONS: NEXT LINE TURNS ON DISPERSAL REQUIREMENT       

C IF (IFLAG(ISP,LX,LY).EQ.0) GO TO 75
 
C..... 

C.....LIGHT FILTER FOR ESTABLISHMENT
 
C..... 


IF (AL.LT.YLI(ISP)) GO TO 75 
C..... 

C.....WINTER TEMPERATURE FILTERS, BUGMANN & SOLOMON 2000

C..... 


IF (TCOLD.LT.WITN(ISP).OR.TCOLD.GT.WITX(ISP)) GO TO 75 
C..... 

C.....DEGREE-DAY FILTER, EVERGREEN OR DECIDUOUS, BUGMANN & SOLOMON 2000 

C..... 


IF (KDECID(ISP).EQ.0) THEN
 
IF (DDE.LT.DDMIN(ISP)) GO TO 75
 
ELSE
 
IF (DDD.LT.DDMIN(ISP)) GO TO 75
 

ENDIF 
C..... 

C.....BROWSING FILTER, 0.167 IS 5/30.0, BUGMANN & SOLOMON 2000

C..... 


IF (RAN(USEED(3)).LT.(FLOAT(IBRW(ISP)-1)*0.167)) GO TO 75 
C..... 

C.....ALLOW A FRACTION OF ELIGIBLE COHORTS IN 

C.....ESTPR=0.1 FOLLOWS FORCLIM (10% OF ELIGIBLE)  

C..... 


ESTPR=0.1
 
IF (RAN(USEED(4)).GE.ESTPR) GO TO 75
 

C..... 

C.....ESTABLISH NEW COHORT  

C..... 


50	 NTOT(IX,IY)=NTOT(IX,IY)+1
IPL=NTOT(IX,IY)
IF(IPL.GE.MT) STOP 'NTOT NEAR MAX FOR BLOCK IN ESTABLISH' 
NCOHS(ISP)=NCOHS(ISP)+1   

DBH(IPL,IX,IY)=SIZE
IF(DBH(IPL,IX,IY).LT.0.0) STOP 'DBH ERROR IN ESTABLISH' 
IAGE(IPL,IX,IY)=0
TLA(IPL,IX,IY)=0.0
TMASS(IPL,IX,IY)=0.0
HTCM(IPL,IX,IY)=0.0
NOGRO(IPL,IX,IY)=0
NOGROL(IPL,IX,IY)=0
NOGROT(IPL,IX,IY)=0
NOGROW(IPL,IX,IY)=0
ISPEC(IPL,IX,IY)=ISP
LOCX(IPL,IX,IY)=LX
LOCY(IPL,IX,IY)=LY
ADP(IPL,IX,IY)=4.6/AGEMX(ISP) 

C..... 
C.....COHORT SIZE IS A FUNCTION OF ESTMX & PATCH SIZE 
C.....OPTIONS: CHOOSE ESTMX OF 0.006, FORCLIM ESTIMATE, BUGMANN 1996  
C.....ALTERNATIVE (2ND EQ) USE SPECIES-SPECIFIC ESTMX VALUES INPUT 
C..... 

NRB=AINT(0.006*CELLAREA+0.5)
C NRB=AINT(ESTMX(ISP)*CELLAREA+0.5) 

ICOHSIZ(IPL,IX,IY)=AINT(RAN(USEED(5))*FLOAT(NRB))+1
C..... 

C.....TO NEXT SPECIES ISP   

C..... 


75 CONTINUE 
C..... 

C.....TO NEXT GRID CELL LX,LY 

C..... 


85 CONTINUE 
C..... 

C.....TALLY ALL COHORTS 

C..... 


NFULL = 0
 
DO 140 I=1,NSPEC


NFULL=NFULL+NCOHS(I)
 
140 CONTINUE
 

WRITE (*,141) KYR,NFULL
 
141 FORMAT(' YR & TOTAL #COHORTS',2I12) 


C..... 
C.....INCREMENT AGES 
C..... 

DO 150 IX=1,NBLKS

DO 150 IY=1,NBLKS

DO 150 I=1,NTOT(IX,IY)

IAGE(I,IX,IY) = IAGE(I,IX,IY)+1
 

150 CONTINUE
 
RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE GROW(IRUN)
C..... 

C.....SUBROUTINE GROW CALCULATES THE ANNUAL STEM DIAMETER INCREMENT 

C.....FOR EACH TREE.  IT IS CALLED BY MAIN ONCE EACH YEAR.  IT IS A 

C.....MODIFICATION OF THE GROW SUBROUTINE IN FORET. 

C..... 


PARAMETER (MT=350,MB=20,MS=20,MC=33) 
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB), 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB), 
>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB), 
>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB), 
>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)
COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS), 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),
>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),
>WITX(MS),IBRW(MS),KDECID(MS)
COMMON/WEATH/T(12),VT(12),RT(13),R(12),VR(12),BS,CW,SLASP,PLAT, 

>AET,DDE,DDD,TCOLD,DRE,DRD
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS 
COMMON/PATCH/IFLAG(MS,MC,MC),SLA(1300,MC,MC),SLAX(1300,MC,MC) 
COMMON NSELCT(MS)
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR 
IF(NFULL.EQ.0) RETURN 
PHI = 1.0 

C..... 

C.....LOOP THROUGH BLOCKS 

C..... 
DO 75 IX=1,NBLKS

DO 75 IY=1,NBLKS
 

C..... 

C.....LOOP THROUGH COHORTS IN BLOCK
 
C.....ESTIMATE DIAMETER GROWTH FOR TREES IN THE ITH FOCAL COHORT 

C..... 


DO 75 I=1,NTOT(IX,IY)

JS = ISPEC(I,IX,IY)
 

C..... 
C.....OPTIONS: NEXT LINE INHIBITS GROWTH OF COHORTS UNTIL THEY REACH 
C.....RECRUITMENT AGE DEFINED BY LTIME, A SPECIES-SPECIFIC TIME LAG
C..... 
C.....RECRUIT SEEDLING COHORTS 
C.....GIVE NEW RECRUITS A DBH OF 1.27 CM 
C..... 
C IF(IAGE(I,IX,IY) .LT. LTIME(JS)) GO TO 75 

IF (DBH(I,IX,IY).EQ.0.) DBH(I,IX,IY)=1.27
C..... 
C.....DETERMINE LOCAL & EXTENDED SHADING LEAF AT AND ABOVE COHORT 
C.....COHORT TREES ARE SELF-SHADING AS IN FORCLIM 
C.....IFHT CAN RANGE FROM 1 TO 1300 
C..... 

SLAS=0.0
 
SLAXS=0.0
 

IFHT=(AMAX1((HTCM(I,IX,IY)-137.0),0.0))/10.+1.
 
DO 71 JH=IFHT,1300


SLAS=SLAS+SLA(JH,LOCX(I,IX,IY),LOCY(I,IX,IY))
 
71 SLAXS=SLAXS+SLAX(JH,LOCX(I,IX,IY),LOCY(I,IX,IY))
 

C..... 

C.....AL IS LIGHT AVAILABLE UNDER LOCAL CELL SHADING LEAF AREA (SLAS) 

C.....FORCLIM LIGHT ATTENUATION CONSTANT (OR LIGHT EXTINCTION COEFF) IS 0.25

C..... 


AL = PHI*EXP(-0.25*SLAS/CELLAREA)

B2 = 2*((HMX(JS)*100.)-137)/DMX(JS)

B3 = B2/(2*DMX(JS))
 

C..... 

C.....CALCULATE MAX TREE VOLUME (GR) & DIAMETER INCREMENT(DNCMX)  

C..... 

C GR = (137.+.25*B2**2/B3)*(0.5*B2/B3) 

C..... 

C.....DIAMETER GROWTH FOLLOWING MOORE 1989, BUGMANN 1996

C..... 


DNCMX = G(JS)*DBH(I,IX,IY)*(1.0-(HTCM(I,IX,IY)/(HMX(JS)*100.))) 
> /(274.+3.0*B2*DBH(I,IX,IY)-4.0*B3*(DBH(I,IX,IY)**2)) 
IF (DNCMX.LT.0.0.OR.DNCMX.GT.5.0) STOP 'DNCMX ERROR IN GROW' 

C..... 

C.....LIGHT RESPONSE CURVES FROM BOTKIN ET AL 1972 

C..... 


GL9 = 2.24*(1.0-EXP(-1.136*(AL-.08)))
 
GL1=(1.0-EXP(-4.64*(AL-.05)))
 

C..... 

C.....MODIFIED FOR NINE TOLERANCE CLASSES FOLLOWING BUGMANN 1996 

C.....FOR HIGH SHADE TOLERANCE ITOL EQUALS 1 

C..... 


SLOPE=(GL9-GL1)/8.0
ALGF=AMAX1((GL1+(FLOAT(ITOL(JS)-1))*SLOPE), 0.0)

C..... 
C.....GROWTH REDUCTION FACTORS FROM BUGMANN & SOLOMON 2000 
C.....DECIDUOUS AND CONIFEROUS SPECIES CALCULATIONS DIFFER 

C..... 
DDGF=AMAX1(0.0,1.0-EXP((DDMIN(JS)-DDE)/750))
IF (KDECID(JS).EQ.1) DDGF= 

> AMAX1(0.0,1.0-EXP((DDMIN(JS)-DDD)/750))
SMGF=SQRT(AMAX1(0.0,1.0-DRE/DRTOL(JS)))

IF (KDECID(JS).EQ.1) SMGF=SQRT(AMAX1(0.0,1.0-DRD/DRTOL(JS))) 
C..... 

C.....OPTIONS: THE SOIL N FACTOR (CONSTANT 0.8 USED IN FORCLIM TESTS  

C.....BY BUSING & SOLOMON 2005) CAN BE REPLACED BY A SNGF CALCULATION  

C.....BASED ON AVAILABLE N ESTIMATED BY A DECOMPOSITION ROUTINE (CF  

C.....BUGMANN 1996)

C..... 


SNGF=0.8 

GF=(ALGF*SMGF*DDGF*SNGF)**0.333

DINC=GF*DNCMX 


C..... 
C.....IS INCREMENT LESS THAN 10% OF OPTIMAL GROWTH (DNCMX), SOLOMON 1986
C.....OR IS INCREMENT LESS THAN A THRESHOLD (0.03 CM, BUGMANN 1994)? 
C.....IF SO, INCREMENT NOGRO, WHICH COUNTS CONSECUTIVE YRS OF SLOW GROWTH
C.....IF LOW LIGHT IS THE PRIMARY STRESS, INCREMENT NOGROL TOO
C.....DO THE SAME FOR CLIMATE STRESSORS 
C..... 

IF (DINC.LT.(0.1*DNCMX).OR.DINC.LT.0.03) THEN 
NOGRO(I,IX,IY) = NOGRO(I,IX,IY)+1 

C..... 

C.....SPECIAL CASE FOR NOGRO TREES LACKING ENVIRONMENTAL STRESS 

C..... 


IF (DINC.GE.(0.1*DNCMX).AND.DINC.LT.0.03) GO TO 60 
C..... 
C.....NOGROL INDICATES LIGHT STRESSED TREES 
C.....NOGROW INDICATES WATER STRESSED TREES 
C.....NOGROT INDICATES COLD STRESSED TREES 
C..... 

NOGROL(I,IX,IY) = NOGROL(I,IX,IY)+1
IF (ALGF.GE.AMIN1(DDGF,SMGF,SNGF)) NOGROL(I,IX,IY) = 0 
NOGROW(I,IX,IY) = NOGROW(I,IX,IY)+1
IF (SMGF.GE.AMIN1(DDGF,ALGF,SNGF)) NOGROW(I,IX,IY) = 0
NOGROT(I,IX,IY) = NOGROT(I,IX,IY)+1
IF (DDGF.GE.AMIN1(ALGF,SMGF,SNGF)) NOGROT(I,IX,IY) = 0  
ELSE 
NOGRO(I,IX,IY) = 0

NOGROL(I,IX,IY) = 0

NOGROT(I,IX,IY) = 0

NOGROW(I,IX,IY) = 0


ENDIF 
C..... 

C.....INCREMENT TREE DIAMETER FOR COHORT
 
C..... 


IF (DINC.LT.0.) WRITE (*,444) JS,HTCM,HMX(JS),GF
 
444 FORMAT (I4,3F12.3)
 
60 DBH(I,IX,IY) = DBH(I,IX,IY)+DINC
 

C..... 

C.....TO NEXT FOCAL COHORT I 

C..... 


75 CONTINUE
 
RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE DISPERSE 
C..... 

C.....SUBROUTINE DISPERSE DETERMINES PROPAGULE AVAILABILITY IN HORIZONTAL 

C.....SPACE. IT IS CALLED BY MAIN ONCE EACH YEAR.  IT CAN BE OMITTED 

C.....BY MAKING A FEW CHANGES OUTLINED IN MAIN. 

C..... 


PARAMETER (MT=350,MB=20,MS=20,MC=33) 
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB),
 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB),

>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB),

>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB),

>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)

COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS),
 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),

>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),

>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),

>WITX(MS),IBRW(MS),KDECID(MS)

COMMON/PATCH/IFLAG(MS,MC,MC),SLA(1300,MC,MC),SLAX(1300,MC,MC)
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
COMMON/SEED/USEED(15)
INTEGER USEED 

C..... 

C.....TWO TYPES OF PROPAGULE INPUT ARE SIMULATED: 

C.....INPUT FROM SOURCES EXOGENOUS IN SPACE & TIME 

C.....INPUT BY LOCAL DISPERSAL FROM ADULT TREES 

C..... 

C.....LOOP THROUGH SPECIES AND CELLS
 
C.....SET PROPAGULE FLAG TO ZERO OR ONE (THE LATTER IF EXOGENOUS SOURCE)

C..... 


DO 15 I=1,NSPEC
DO 15 MX=1,IWIDTH

DO 15 MY=1,IWIDTH
 
IF (RAN(USEED(6)).LT.(DISP(I))) THEN
 
IFLAG(I,MX,MY)=1

ELSE 
IFLAG(I,MX,MY)=0
ENDIF


 15 CONTINUE
 
C..... 

C.....LOOP THOUGH BLOCKS 

C..... 


DO 130 KKX=1,NBLKS
 
DO 130 KKY=1,NBLKS
 

C..... 

C.....LOOP THROUGH COHORTS IN EACH BLOCK
 
C.....CHECK FOR ADULT-SIZED TREES & DISPERSE PROPAGULES FROM THEM 

C..... 


DO 130 KK=1,NTOT(KKX,KKY)
ISP=ISPEC(KK,KKX,KKY)

IF(DBH(KK,KKX,KKY).LT.DREPR(ISP)) GO TO 130 
C..... 

C.....KS IS THE NUMBER OF DISPERSAL EVENTS PER ADULT PER YR
 
C.....KS IS NOW SET TO ONE PER ADULT PER YR 

C..... 


16 KS=ICOHSIZ(KK,KKX,KKY) 
C..... 

C.....LOOP THROUGH DISPERSAL EVENTS
 
C..... 


DO 75 IJSE=1,KS 

C..... 
C.....DISPERSE PROPAGULE FROM ADULT DDIST AT DIRECTION DTHET 
C.....CALCULATE THE CELL COORDINATES (LX,LY) OF A PROPAGULE
C.....IF PROPAGULE LANDS MORE THAN IWIDTH OFF TRACT GO TO NEXT ADULT 
C..... 

DDIST=-DISD(ISP)*ALOG(RAN(USEED(7)))
 
DTHET=RAN(USEED(8))*6.283
 

C..... 

C.....CONVERT REAL DISPERSAL COORDINATES TO NEAREST INTEGER
 
C..... 


LX=NINT(SIN(DTHET)*DDIST)+LOCX(KK,KKX,KKY)
 
IF(LX.LT.1) LX=LX+IWIDTH

IF(LX.GT.IWIDTH) LX=LX-IWIDTH

IF(LX.GT.IWIDTH) GO TO 75
 
IF(LX.LT.1) GO TO 75
 
LY=NINT(COS(DTHET)*DDIST)+LOCY(KK,KKX,KKY)
 
IF(LY.LT.1) LY=LY+IWIDTH

IF(LY.GT.IWIDTH) LY=LY-IWIDTH

IF(LY.GT.IWIDTH) GO TO 75 

IF (LY.LT.1) GO TO 75
 

C..... 

C.....FLAG CELL FOR SPECIES TO REGENERATE VIA ENDOGENOUS SOURCES (IFLAG>1)

C..... 


IFLAG(ISP,LX,LY)=2
C..... 

C.....TO NEXT DISPERSAL EVENT IJSE
 
C..... 


75 CONTINUE 
C..... 

C.....TO NEXT POTENTIAL SEED DISPERSING ADULT COHORT KK
 
C..... 


130 CONTINUE

 150 RETURN
 

END
 
C..... 
C..... 
C..... 

SUBROUTINE CANOPY 
C..... 

C.....SUBROUTINE CANOPY UPDATES MASS, FOLIAGE, HEIGHT, AND CANOPY STRUCTURE

C.....IT IS CALLED FROM MAIN ONCE EACH YEAR.  ONE OF TWO EQUATIONS FOR 

C.....ESTIMATION OF TREE HEIGHT SHOULD BE SELECTED.   

C..... 


PARAMETER (MT=350,MB=20,MS=20,MC=33) 
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB),
 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB),
 
>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB),
 
>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB),
 
>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)

COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS),
 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),

>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),

>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),

>WITX(MS),IBRW(MS),KDECID(MS)

COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS 
COMMON/PATCH/IFLAG(MS,MC,MC),SLA(1300,MC,MC),SLAX(1300,MC,MC) 
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR 
COMMON/SEED/USEED(15)
INTEGER USEED 

C..... 

C.....LOOP THROUGH BLOCKS AND COHORTS IN EACH BLOCK 
C.....UPDATE FOLIAGE, BIOMASS, AND HEIGHT FOR EACH COHORT
C..... 

DO 10 KX=1,NBLKS
 
DO 10 KY=1,NBLKS

DO 10 K=1,NTOT(KX,KY)


J=ISPEC(K,KX,KY)
 
C..... 

C.....TREE FOLIAGE WET WEIGHT & COHORT LEAF AREA  

C..... 


FOLWET=EXP(FC1(J))*DBH(K,KX,KY)**FC2(J) 
C FOLDRY=FOLWET*0.45 
C IF(KDECID(J).EQ.1) FOLDRY=FOLWET*0.35  

TLA(K,KX,KY)=FLOAT(ICOHSIZ(K,KX,KY))*FC0(J)*FOLWET 
C..... 

C.....COHORT ABOVEGROUND BIOMASS ESTIMATE FROM FORET, SHUGART & WEST 1977

C.....NOTE: THIS ESTIMATE CAN BE HIGH FOR SOFTWOODS 

C..... 


TMASS(K,KX,KY)=ICOHSIZ(K,KX,KY)*0.1193*DBH(K,KX,KY)**2.393 
C..... 
C.....OPTIONS: PARABOLIC OR ASYMPTOTIC TREE HEIGHT ESTIMATION 
C.....SELECT ASYMPTOTIC HEIGHT ESTIMATE, RICHARDS 1959
C.....NOTE SETTING HMX FROM HC0 IS POSSIBLE HERE 
C.....THIS MAY REQUIRE GROWTH EQ ADJUSTMENTS  
C..... 
C HMX(J)=HC0(J)+1.37       
C HTCM(K,KX,KY)=100.*(HMX(J)* 
C > (1-EXP(HC1(J)*DBH(K,KX,KY)))**HC2(J))

B2 = 2.0*((HMX(J)*100.)-137.0)/DMX(J)
B3 = B2/(2.0*DMX(J))

C..... 

C.....OR SELECT PARABOLIC TREE HEIGHT ESTIMATE FROM BOTKIN ET AL 1972 

C.....FOLLOWS KER & SMITH 1955 

C..... 


HTCM(K,KX,KY)=137.0+B2*DBH(K,KX,KY)-B3*DBH(K,KX,KY)**2 
10 CONTINUE 

C..... 

C.....LOOP THROUGH CELLS AND VERTICAL STRATA IN EACH CELL
 
C..... 


DO 24 LX=1,IWIDTH

DO 24 LY=1,IWIDTH


DO 24 JH=1,1300
 
C..... 

C.....CLEAR VERTICAL CANOPY PROFILE ARRAYS FOR EACH CELL
 
C..... 


SLA(JH,LX,LY)=0.0
 
24 SLAX(JH,LX,LY)=0.0


C..... 

C.....DETERMINE LOCAL & EXTENDED VERTICAL LEAF AREA PROFILE FOR CELLS 

C.....KRADIUS IS THE NEIGHBORHOOD RADIUS FOR EXTENDED PATCH EFFECTS 

C.....KRADIUS SHOULD NOT EXCEED KWIDTH, UNITS ARE SQRT(CELLAREA)  

C.....SET KRADIUS GE 1 FOR EFFECTS OF NEIGHBORING CELLS (IN SLAX) 

C..... 


KRADIUS=1 

KRSQ=KRADIUS**2.0


IF (KRSQ.LE.0) STOP 'KRADIUS ERROR IN CANOPY' 

KRAD=KRADIUS
 
NRAD=-KRAD
 

C..... 

C.....LOOP THROUGH CELLS, COORDINATES LX,LY 

C..... 
DO 75 LX=1,IWIDTH
DO 75 LY=1,IWIDTH

C..... 
C.....DETERMINE WHICH BLOCK IX,IY CELL IS IN 
C..... 

IX=((LX-1)/KWIDTH)+1
IY=((LY-1)/KWIDTH)+1
IF(IX.LT.1.OR.IY.LT.1) STOP 'LOW BLOCK ERROR IN CANOPY' 
IF(IX.GT.NBLKS.OR.IY.GT.NBLKS) STOP 'HIGH BLOCK ERROR IN CANOPY' 

C..... 
C.....LOOP THROUGH NEIGHBORHOOD BLOCKS 
C.....NEIGHBORHOOD BLOCKS ARE JX,KY, ADJUSTMENT MADE FOR BLOCKS AT TRACT 
C.....EDGE, ACCUMULATE LEAF AREA SUM (SLA) OF COHORTS IN THE SPECIFIED  
C.....RANGE OF THE CELL LX,LY 
C..... 

DO 41 J=1,3
IXSHIFT=0 
JX=IX-2+J 
IF(JX.LT.1) THEN 
JX=JX+NBLKS 
IXSHIFT=-IWIDTH 
ENDIF 
IF(JX.GT.NBLKS) THEN 
JX=JX-NBLKS 
IXSHIFT=IWIDTH 
ENDIF 

DO 41 K=1,3
IYSHIFT=0 
KY=IY-2+K 
IF(KY.LT.1) THEN 
KY=KY+NBLKS 
IYSHIFT=-IWIDTH 
ENDIF 
IF(KY.GT.NBLKS) THEN 
KY=KY-NBLKS 
IYSHIFT=IWIDTH 
ENDIF 
IF(NTOT(JX,KY).EQ.0) GO TO 41 

C..... 
C.....LOOP THROUGH COHORTS IN NEIGHBORHOOD BLOCK, FIND PROXIMATE COHORTS
C..... 

DO 40 L=1,NTOT(JX,KY)
IDX=LOCX(L,JX,KY)-LX+IXSHIFT
IF(IDX.GT.KRAD.OR.IDX.LT.NRAD) GO TO 40 
IDY=LOCY(L,JX,KY)-LY+IYSHIFT
IF(IDY.GT.KRAD.OR.IDY.LT.NRAD) GO TO 40 
IDISQ=IDX**2+IDY**2
IF(IDISQ.GT.KRSQ) GO TO 40 

DIST=SQRT(FLOAT(IDISQ))
IF(IAGE(L,JX,KY).GT.0) THEN 

C..... 
C.....ADD PROXIMATE COHORTS TO CANOPY HEIGHT PROFILE IN 10 CM STRATA 
C.....FIND HEIGHT STRATUM IHT OF NEIGHBOR COHORT & ADD ITS LEAF AREA TLA 
C..... 
C.....LINKAGES EQ: IHT=(HTCM(L,JX,KY)/10.)+1. 
C.....FORCLIM EQ FOR C ARRAY, PLUS 1. FOR FORTRAN ARRAY:
C..... 

IHT=((HTCM(L,JX,KY)-137.0)/10.)+1.       
IF (IHT.LT.1.AND.DBH(L,JX,KY).GT.20) WRITE (*,58)HTCM(L,JX,KY), 

> DBH(L,JX,KY),ISPEC(L,JX,KY),IAGE(L,JX,KY) 
58 FORMAT('SUBR CANOPY: ZERO OR NEGATIVE IHT FOR ADULT',2F12.4,2I10)


IF (IHT.GT.1300) STOP 'IHT EXCEEDED LIMIT IN CANOPY'
 
IF (IHT.LT.1) IHT=0
 

C..... 

C.....ADD LEAF AREAS INTO CANOPY PROFILE OF CELL 

C..... 


IF(IDX.EQ.0.AND.IDY.EQ.0.AND.IHT.GE.1) SLA(IHT,LX,LY)= 
> SLA(IHT,LX,LY)+TLA(L,JX,KY) 

C..... 
C.....DETERMINE LEAF AREA PROFILE OF EXTENDED NEIGHBORHOOD 
C.....IF COHORT IS IN OR NEAR FOCAL CELL ADD LEAF AREAS TO PROFILE 
C.....DIVIDE BY NUMBER OF PATCHES COMPRISING NEIGHBORHOOD TO OBTAIN 
C.....MEAN SLA PER PATCH 
C..... 

IF (IDISQ.LT.20.AND.IHT.GE.1) 
> SLAX(IHT,LX,LY)=SLAX(IHT,LX,LY)+(TLA(L,JX,KY)/ 
> (3.14*KRSQ))

ENDIF

 40 CONTINUE
 

C..... 

C.....TO NEXT NEIGHBORHOOD BLOCK J,K

C..... 


41 CONTINUE 
C..... 

C.....TO NEXT FOCAL CELL LX,LY 

C..... 


75 CONTINUE
 
RETURN
 
END 


C..... 
C..... 
C..... 

SUBROUTINE TEMPE 
C..... 
C.....SUBROUTINE TEMPE CALCULATES DEGREE DAYS FOR FULL YEAR AND GROWING 
C.....SEASON, AND FINDS MINIMUM WINTER TEMPERATURES.  MONTHLY TEMPERATURES 
C.....NORMALLY DISTRIBUTED AROUND SPECIFIED MEANS AND STD DEVS ARE THE  
C.....BASIS FOR THE CALCULATIONS.  TEMPERATURES ARE GENERATED BY SUBROUTINE 
C.....GGNORD USING RANDOM NUMBERS.  CLIMATE CHANGE CAN BE MODELED BY LINEAR 
C.....INTERPOLATION BETWEEN YEARS OF DIFFERENT CLIMATES BY SUBROUTINE LININT. 
C.....SUBROUTINE TEMPE IS CALLED BY MAIN ONCE EACH YEAR.  IT IS A MODIFICATION 
C.....OF THE TEMPE SUBROUTINE IN THE LINKAGES MODEL. 
C..... 

PARAMETER (MT=350,MB=20,MS=20)
COMMON/WEATH/T(12),VT(12),RT(13),R(12),VR(12),BS,CW,SLASP,PLAT, 

>AET,DDE,DDD,TCOLD,DRE,DRD

COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR

COMMON/SEED/USEED(15)

DIMENSION DAYS(12),Z(2)
 
INTEGER USEED
 
DATA DAYS/31.,28.,31.,30.,31.,30.,31.,31.,30.,31.,30.,31./
 
DATA NCT/0/
 

C..... 
YR = FLOAT(KYR)

C..... 

C.....INITIALIZE PREVIOUS DEC TEMP AT ARBITRARY HIGH VALUE, FORCLIM2.9 

C..... 


IF(KYR.EQ.1) RT(13)=100. 
C..... 

http:DAYS/31.,28.,31.,30.,31.,30.,31.,31.,30.,31.,30.,31

C.....DDBASE IS TEMPERATURE (C) ABOVE WHICH DEGREE DAYS ARE
C.....COUNTED; FORCLIM2.9 VALUE IS 5.5, LINKAGES 5.56
C..... 

DDBASE = 5.5 
C..... 

C.....INITIALIZE DEGREE DAYS AND MONTHLY TEMPERATURES USED TO
 
C.....CALCULATE THEM (IN ARRAY RT)

C..... 


DDE = 0.
 
DDD = 0. 

DO 3 I=1,12
 

3 RT(I) = 0.
C..... 

C.....CALL SUBROUTINE LININT FOR LINEAR INTERPOLATIONS OF MONTHLY 

C.....TEMPERATURES AND STD DEVS BETWEEN YEARS OF DIFFERENT CLIMATE. 

C..... 


CALL LININT(T,VT,YR,1)
C..... 

C.....LOOP THROUGH MONTHS 

C..... 


DO 10 I=1,12

NCT = NCT+1
 
IF(NCT .EQ. 2) GO TO 36
 

C..... 

C.....CALL GGNORD TO PROVIDE NORMALLY DISTRIBUTED RANDOM NUMBERS 

C..... 


CALL GGNORD(USEED(9),USEED(10),Z)

GO TO 38


 36 Z(1) = Z(2)

NCT = 0


 38 CONTINUE 
C..... 

C.....CALCULATE MONTHLY TEMPERATURES AS THE INTERPOLATED MEAN +/- 

C.....NORMALLY DISTRIBUTED RANDOM NUMBER TIMES INTERPOLATED STD DEV 

C..... 


RT(I) = T(I)+VT(I)*Z(1) 
C..... 
C.....BUGMANN & SOLOMON 2000 MINIMUM WINTER TEMPERATURE FROM DEC, JAN OR FEB
C.....FIND COLDEST WINTER MONTHLY MEAN TEMP FOR YEAR & FOR PREV DEC 
C.....RT(13) HOLDS PREV DEC 
C..... 

TCOLD=AMIN1(RT(1),RT(2),RT(13))
C IF(RT(I) .LE. DDBASE) GO TO 10
 
C..... 

C.....CALCULATE DEGREE DAY CORRECTION FACTOR, FROM FORCLIM2.9 C# 

C..... 


CORR=-31.8+2.377*RT(I)

IF(RT(I).LE.5.5) CORR=8.52*10**(0.165*RT(I))

IF(RT(I).GT.5.5.AND.RT(I).LE.15.5) CORR=187.2*10**(-0.0908*RT(I))
 

C..... 
C.....SUM DEGREE DAYS FOR CONSECUTIVE MONTHS THROUGH FULL YEAR 
C.....ADD FORCLIM CORRECTION FACTOR CORR, BUGMANN 1996
C.....OPTIONS: DAYS(I) IS LINKAGES CALCULATION 
C.....30.5 DAYS IS FORCLIM2.9 CALCULATION 
C..... 
C DD=AMAX1(RT(I)-DDBASE,0.0)*DAYS(I)+CORR 

DD=AMAX1(RT(I)-DDBASE,0.0)*30.5+CORR

DDE = DDE+DD
 

C..... 

C.....ALSO SUM DEGREE DAYS ONLY WHEN DECIDUOUS TREES ARE IN LEAF 

C.....CF BUGMANN & SOLOMON 2000 
C..... 

IF (I.GT.3.AND.I.LT.11) DDD = DDD+DD
WRITE(*,673)I,RT(I),CORR,DD,DDE,DDD 

673 FORMAT(I3,5F8.2)
10 CONTINUE 

WRITE(*,674)DDE,DDD
674 FORMAT(' DDE DDD',2F12.3)

C..... 
C.....STORE DEC TEMP FOR USE NEXT YEAR 
C..... 

RT(13)=RT(12)
RETURN 
END 

C..... 
C..... 
C..... 

SUBROUTINE WATER 
C..... 
C.....SUBROUTINE WATER CALCULATES THE FRACTION OF THE GROWING SEASON  
C.....AND THE DECIDUOUS SEASON WITH UNFAVORABLE SOIL MOISTURE FOR GROWTH 
C.....AND ACTUAL EVAPOTRANSPIRATION (AET) USED TO DETERMINE DECAY RATES.
C.....THE SUBROUTINE ESTIMATES THORNTHWAITE PET & DROUGHT INDICES. 
C.....TEMPERATURES ARE PROVIDED BY SUBROUTINE TEMPE.  MONTHLY 
C.....PRECIPITATION IS GENERATED AS TEMPERATURES ARE IN TEMPE.  THE 
C.....SUBROUTINE IS CALLED FROM MAIN ONCE EACH YEAR.  IT CONTAINS  
C.....ELEMENTS OF THE SOIL MOISTURE ROUTINES IN LINKAGES AND IN 
C.....FORCLIM VERSION 2.9. 
C..... 
C..... 
C..... 
C..... 
C..... 

REQUIRED DATA READ BY SUBROUTINE READSITE: 
RT CONTAINS MEAN MONTHLY (JAN-DEC) TEMPERATURES IN DEG C 
R IS MEAN MONTHLY (JAN-DEC) RAINFALL IN CM  
VR IS STANDARD DEVIATION ABOUT MEAN MONTHLY RAINFALL 

C..... 
C..... 

PLAT IS LATITUDE OF SITE (DEGREES NORTH) 

PARAMETER (MT=350,MB=20,MS=20)
COMMON/WEATH/T(12),VT(12),RT(13),R(12),VR(12),BS,CW,SLASP,PLAT, 

>AET,DDE,DDD,TCOLD,DRE,DRD
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
COMMON/SEED/USEED(15)
DIMENSION Z(2),CLAT(12,26)

C DIMENSION Z(2),CLAT(12,26),DLAT(12,6) 
DIMENSION DAYS(12),DR(12),SM(13),AK(12),BK(12) 

C EQUIVALENCE (CLAT(1,21),DLAT(1,1))
INTEGER USEED 

C..... 

C.....POTENTIAL EVAPOTRANSPIRATION FACTORS FOR SUN ANGLE & DAY LENGTH 

C.....MONTHLY CORRECTION FACTORS FOR 25-50 DEGREES LATITUDE NORTH 

C.....DAYS(K) IS NUMBER OF DAYS BETWEEN MID-MONTH K-1 AND K 

C.....OPTIONS: CORRECTIONS CAN BE OBTAINED BY ESTIMATION (BUGMANN 1996) 

C.....OR FROM TABLES (PASTOR & POST 1985)

C..... 

C.....ESTIMATION FACTORS AK & BK USED IN MONTH LOOP, BUGMANN 1994 

C..... 


DATA AK/1.1226,.9859,1.0454,.9708,.9605,.9185,.9669,.9892,.99, 
> 1.06,1.0815,1.1444/
DATA BK/-7.3094E-3,-3.8701E-3,-4.9231E-4,3.5179E-3,7.1453E-3, 

> 8.4718E-3,7.641E-3,4.9436E-3,1.2E-3,-2.6256E-3,-6.3692E-3, 
> -8.6598E-3/ 

C..... 

C.....ALTERNATIVE METHOD USING TABULAR VALUES 

C.....PASTOR & POST 1985, TABLE OF CORRECTION FACTORS

C..... 

C DATA DLAT/.80,.81,1.02,1.13,1.28,1.29,1.31,1.21,1.04,.94,.79,.75,
 
C 6 .79,.81,1.02,1.13,1.29,1.31,1.32,1.22,1.04,.94,.79,.74, 

C 7 .77,.80,1.02,1.14,1.30,1.32,1.32,1.22,1.04,.93,.78,.73, 

C 8 .76,.80,1.02,1.14,1.31,1.33,1.34,1.23,1.05,.93,.77,.72, 

C 9 .75,.79,1.02,1.14,1.32,1.34,1.35,1.24,1.05,.93,.76,.71, 

C * .74,.78,1.02,1.15,1.33,1.36,1.37,1.25,1.06,.92,.76,.70/ 


DATA CLAT/.93,.89,1.03,1.06,1.15,1.14,1.17,1.12,1.02,.99,.91,.91, 
6 .92,.88,1.03,1.06,1.15,1.15,1.17,1.12,1.02,.99,.91,.91, 
7 .92,.88,1.03,1.07,1.16,1.15,1.18,1.13,1.02,.99,.90,.90, 
8 .91,.88,1.03,1.07,1.16,1.16,1.18,1.13,1.02,. 8,.90,. 0, 
9 .91,.87,1.03,1.07,1.17,1.16,1.19,1.13,1.03,.98,.90,.89, 
* .90,.87,1.03,1.08,1.18,1.17,1.20,1.14,1.03,.98,.89,.88, 
1 .90,.87,1.03,1.08,1.18,1.18,1.20,1.14,1.03,.98,.89,.88, 
2 .89,.86,1.03,1.08,1.19,1.19,1.21,1.15,1.03,.98,.88,.87, 
3 .88,.86,1.03,1.09,1.19,1.20,1.22,1.15,1.03,.97,.88,.86, 
4 .88,.85,1.03,1.09,1.20,1.20,1.22,1.16,1.03,.97,.87,.86, 
5 .87,.85,1.03,1.09,1.21,1.21,1.23,1.16,1.03,.97,.86,.85, 
6 .87,.85,1.03,1.10,1.21,1.22,1.24,1.16,1.03,.97,.86,.84, 
7 .86,.84,1.03,1.10,1.22,1.23,1.25,1.17,1.03,.97,.85,.83, 
8 .85,.84,1.03,1.10,1.23,1.24,1.25,1.17,1.04,.96,.84,.83, 
9 .85,.84,1.03,1.11,1.23,1.24,1.26,1.18,1.04,.96,.84,.82, 
* .84,.83,1.03,1.11,1.24,1.25,1.27,1.18,1.04,.96,.83,.81, 
1 .83,.83,1.03,1.11,1.25,1.26,1.27,1.19,1.04,.96,.82,.80, 
2 .82,.83,1.03,1.12,1.26,1.27,1.28,1.19,1.04,.95,.82,.79, 
3 .81,.82,1.02,1.12,1.26,1.28,1.29,1.20,1.04,.95,.81,.77, 
4 .81,.82,1.02,1.13,1.27,1.29,1.30,1.20,1.04,.95,.80,.76, 
5 .80,.81,1.02,1.13,1.28,1.29,1.31,1.21,1.04,.94,.79,.75, 
6 .79,.81,1.02,1.13,1.29,1.31,1.32,1.22,1.04,.94,.79,.74, 
7 .77,.80,1.02,1.14,1.30,1.32,1.32,1.22,1.04,.93,.78,.73, 
8 .76,.80,1.02,1.14,1.31,1.33,1.34,1.23,1.05,.93,.77,.72, 
9 .75,.79,1.02,1.14,1.32,1.34,1.35,1.24,1.05,.93,.76,.71, 
* 	 .74,.78,1.02,1.15,1.33,1.36,1.37,1.25,1.06,.92,.76,.70/ 
DATA DAYS/31.,28.,31.,30.,31.,30.,31.,31.,30.,31.,30.,30./ 
DATA NCT/0/

C..... 

C.....ADJUST LATITUDE POINTER 

C..... 


LAT=(PLAT+.5)-24
C..... 

C.....EXCEPTION FOR EXTREME LATITUDES, FOLLOWING BUGMANN 1994

C..... 


IF (PLAT.GT.50.OR.PLAT.LT.-50) PLAT=50 
C..... 
C.....INITIALIZE SOIL MOISTURE AT CAPACITY, SLOPE ASPECT FACTORS FOR PET
C.....FORCLIM2.9 SLASP (RANGE -2 TO 2) IS NEGATIVE ON N FACING SLOPES 
C.....IN THE NORTHERN HEMISPHERE 
C..... 

IF (KYR.EQ.1) THEN 

DO 43 I=1,13
 

43 SM(I)=BS

ENDIF 

C..... 

C.....CARRY OVER DEC SOIL MOISTURE TO NEXT JAN 

C..... 


SM(1)=SM(13)

IF(SLASP.GT.0.0) THEN
 
PMOD=1.0+SLASP*0.125

ELSE 
PMOD=1.0+SLASP*0.063 
ENDIF 

C..... 
C.....INITIALIZE THORNTHWAITE PARAMETERS 
C.....TE = TEMPERATURE EFFICIENCY 
C.....A = EXPONENT OF EVAPOTRANSPIRATION FUNCTION 
C.....PET = POTENTIAL EVAPOTRANSPIRATION 
C.....AET = ACTUAL EVAPOTRANSPIRATION 
C.....NE = NUMBER OF MONTHS TALLIED FOR EVERGREEN DROUGHT INDEX 
C.....ND = NUMBER OF MONTHS TALLIED FOR DECIDUOUS DROUGHT INDEX 
C.....SUME = ANNUAL SUM OF MONTHLY DROUGHT INDEX FOR EVERGREEN 
C.....SUMD = ANNUAL SUM OF MONTHLY DROUGHT INDEX FOR DECIDUOUS 
C.....DRE,DRD = MEAN VALUES USED AS DROUGHT INDICES FOR YEAR 
C..... 

PET=0.0 

AET=0.0
 
AAET=0.0
 
APET=0.0
 
TE=0.
 
NE=0
 
SUME=0.0
 
ND=0
 
SUMD=0.0
 

C..... 

C.....LOOP THROUGH MONTHS 

C.....CALCULATE TEMPERATURE EFFICIENCY BY MONTH, SUM FOR YEAR

C..... 


DO 10 K=1,12
 
10 TE=TE+(AMAX1(0.0,.2*RT(K)))**1.514
 

C..... 

C.....CALCULATE EVAPOTRANSPIRATION FUNCTION EXPONENT FOR YEAR
 
C..... 


A=0.000001*((.675*TE**3)+(-77.1*TE**2)+(17920.0*TE)+492390.0)
C..... 

C.....CALL LININT TO INTERPOLATE MONTHLY PRECIP AND STD DEV
 
C.....BETWEEN YEARS OF DIFFERENT CLIMATE
 
C..... 


YR = FLOAT(KYR)
 
CALL LININT(R,VR,YR,2)

WRITE(*,767)
 

767 FORMAT(' MO RAIN PI PS D SM S E SMNEW DR') 
C..... 

C.....LOOP THROUGH MONTHS 

C.....CALCULATE ANNUAL WATER BALANCE BY MONTH, SUMS FOR YEAR

C..... 


DO 50 K=1,12
C..... 

C.....ESTIMATE OF PET LATITUDE CORRECTION FACTOR, BUGMANN 1994

C..... 


PTRLAT=AK(K)+BK(K)*PLAT

NCT=NCT+1
 
IF(NCT.EQ.2) GO TO 36
 

C..... 

C.....CALL GGNORD FOR NORMALLY DISTRIBUTED RANDOM NUMBER
 
C..... 


CALL GGNORD(USEED(11),USEED(12),Z)

GO TO 38


 36 Z(1)=Z(2)

NCT=0

C..... 

C.....CALCULATE MONTHLY PRECIP AS THE INTERPOLATED MEAN +/- NORMALLY 

C.....DISTRIBUTED RANDOM NUMBER TIMES THE INTERPOLATED STD DEV
 
C..... 


38 	 RAIN=R(K)+VR(K)*Z(1)

IF(RAIN.LT.0.0) RAIN=0.0

RAIN=AMAX1(0.0,RAIN)
 

C..... 

C.....CALCULATE POTENTIAL EVAPOTRANSPIRATION (PET) 

C.....THORNTHWAITE & MATHER 1957
 
C..... 

C.....OPTIONS: USE TABULAR LAT VALUE (1ST EQ) OR ESTIMATE OF BUGMANN (2ND EQ)

C..... 

C PET=PMOD*1.6*((10.0*AMAX1(RT(K),0.0)/TE)**A)*CLAT(K,LAT) 


PET=PMOD*1.6*((10.0*AMAX1(RT(K),0.0)/TE)**A)*PTRLAT 
C..... 

C.....BUCKET SIZE (BS) MODEL OF BUGMANN & CRAMER 1998, FORCLIM2.9 

C.....CALCULATE PRECIP INTERCEPTED (PI), FRACTION SET AT 0.3, INFILTRATED (PS), 

C.....DEMAND (D), SUPPLY (S), TRANSPIRATION (E), MONTHLY DROUGHT INDEX (DR) 

C.....RISES AS THE RATIO OF EVAPOTRANSPIRATION TO DEMAND FALLS
 
C..... 


PI=AMIN1(0.3*RAIN,PET)

PS=RAIN-PI
 
D=PET-PI
 
S=CW*SM(K)/BS

E=AMIN1(S,D)

AET=E+PI
 
SMNEW=AMAX1(AMIN1(SM(K)+PS-E,BS),0.0)
 

C..... 

C.....CALCULATE MONTHLY DROUGHT INDEX DR(K) 

C.....ARRAY STORAGE IS OPTIONAL
 
C..... 


IF (D.NE.0.0)THEN 

DR(K)=1.0-E/D

ELSE 
DR(K)=0.0
ENDIF
 
WRITE (*,432) K,RAIN,PI,PS,D,SM(K),S,E,SMNEW,DR(K)
 

432 FORMAT (I3,9F7.3)
 
C..... 

C.....SUM MONTHLY EVAPOTRANSPIRATION VALUES 

C..... 


APET=APET+PET
 
AAET=AAET+AET
 

C..... 

C.....CARRY OVER SOIL MOISTURE TO NEXT MONTH 

C..... 


SM(K+1)=SMNEW
C..... 

C.....SUM DROUGHT INDICES OVER MONTHS WITH SUFFICIENT DEGREE DAYS 

C..... 


IF (RT(K).LE.5.5) GO TO 50 

NE=NE+1
 
SUME=SUME+DR(K)


C..... 

C.....DECIDUOUS GROWING SEASON SUM
 
C..... 


IF (K.GT.3.AND.K.LT.11)THEN
 
ND=ND+1
 
SUMD=SUMD+DR(K)

ENDIF

 50 CONTINUE
 

C..... 

C.....EXIT MONTHLY LOOP AND TAKE ANNUAL SUMS 

C.....THEN CALCULATE ANNUAL DROUGHT INDICES 

C..... 


IF (NE.GT.0)THEN 

DRE=SUME/FLOAT(NE)
 
ELSE 

DRE=0.0 
ENDIF
 
IF (ND.GT.0)THEN 

DRD=SUMD/FLOAT(ND)
 
ELSE 

DRD=0.0 
ENDIF
 
WRITE(*,521)SUME,DRE,SUMD,DRD
 

521 FORMAT(' SUME DRE SUMD DRD',4F9.3)
 
C..... 

C.....CONVERT ANNUAL SUMS FROM CM TO MM, FORCLIM2.9 

C..... 


APET=APET*10.
 
AAET=AAET*10.
 
RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE OUTPUT(IS,IRUN)
C..... 

C.....SUBROUTINE OUTPUT WRITES DATA TO A FILE.  IT IS CALLED BY MAIN 

C.....AT A SPECIFIED INTERVAL OF SIMULATION YEARS. 

C..... 


PARAMETER (MT=350,MB=20,MS=20)
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB), 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB), 
>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB), 
>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB), 
>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)
COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS), 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS), 
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS), 
>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS), 
>WITX(MS),IBRW(MS),KDECID(MS)
COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
COMMON/WEATH/T(12),VT(12),RT(13),R(12),VR(12),BS,CW,SLASP,PLAT, 

>AET,DDE,DDD,TCOLD,DRE,DRD

IF (NFULL.EQ.0) RETURN
 
DO 200 JX=1,NBLKS

DO 200 JY=1,NBLKS

DO 200 J=1,NTOT(JX,JY)


HTM=HTCM(J,JX,JY)/100.
 
C..... 

C.....OUTPUT DATA FILE, ONE LINE PER COHORT 

C..... 


IF(DBH(J,JX,JY).GT.1.) WRITE(6,100)IS,IRUN,KYR,
> ISPEC(J,JX,JY),ICOHSIZ(J,JX,JY),DBH(J,JX,JY),HTM, 
> TLA(J,JX,JY),IAGE(J,JX,JY),LOCX(J,JX,JY),LOCY(J,JX,JY), 
> NOGRO(J,JX,JY),NOGROL(J,JX,JY),NOGROW(J,JX,JY), 

> NOGROT(J,JX,JY),DRE,DRD,DDE,DDD,TCOLD,TOTAR   

100 FORMAT(2I3,I5,2I4,2F6.1,F8.1,I5,2I4,4I3,2F5.2,2F8.1,2F6.1)
 
200 CONTINUE
 

RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE GGNORD(NSEED1,NSEED2,Z) 
C..... 

C.....SUBROUTINE GGNORD CALCULATES NORMALLY DISTRIBUTED RANDOM NUMBERS 

C.....FROM UNIFORM RANDOM NUMBERS SUPPLIED BY RAN. IT IS CALLED EACH MONTH
 
C.....FROM SUBROUTINES TEMPE AND WATER.
 
C..... 


DIMENSION Z(1)
 
DATA PI2/0.62831853E01/

K = 0
 

C..... 

C.....GET RANDOM NUMBERS 

C..... 


A1 = RAN(NSEED1)
 
A2 = RAN(NSEED2)
 
K = K+1
 

C..... 

C.....CALCULATE NORMALLY DISTRIBUTED RANDOM NUMBERS.
 
C.....EMSHOFF & SISSON 1970 

C..... 


Z(K) = SQRT(-.2E01*ALOG(A1))*SIN(PI2*A2)
 
K = K+1
 
Z(K) = SQRT(-0.2E01*ALOG(A1))*COS(PI2*A2)
 
RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE LININT(P1,P2,XX,NTYPE) 
C..... 
C.....SUBROUTINE LININT INTERPOLATES MONTHLY TEMPERATURES, PRECIPITATION
C.....AND THEIR STD DEVS FOR ALL YEARS BRACKETED BY TWO YEARS OF 
C.....DIFFERENT CLIMATES. THESE YEARS ARE SUPPLIED IN ARRAY X.  IT IS 
C.....CALLED BY SUBROUTINES TEMPE & WATER.  THIS SUBROUTINE IS A  
C.....MODIFICATION OF THE LININT SUBROUTINE IN LINKAGES. 
C..... 

DIMENSION P1(12),P2(12)
 
COMMON/INTERP/IPOLAT,X(10)
 
COMMON /LINEAR/TSAV(45,12),VTSAV(45,12),RSAV(45,12),VRSAV(45,12)

NPTS = IPOLAT
 
NPT1 = NPTS-1
 

C..... 
C.....FIND YEARS BETWEEN WHICH LINEAR INTERPOLATIONS SHOULD BE MADE 
C.....XX IS CURRENT YEAR. X(I) AND X(I+1) ARE BRACKETING YEARS
C.....SPECIFIED IN ARRAY X 
C..... 

DO 250 I=1,NPT1
IF(XX .GT. X(I) .AND. XX .LE. X(I+1)) GO TO 300


 250 CONTINUE

 300 CONTINUE
 

DO 500 K=1,12
C..... 

C.....IF TEMPE CALLS LININT, NTYPE = 1

C.....IF WATER CALLS LININT, NTYPE = 2
C..... 

IF(NTYPE .EQ. 2) GO TO 400 
C..... 

C.....INTERPOLATE MEAN MONTHLY TEMPERATURES (C) BETWEEN YEARS

C.....OF DIFFERENT CLIMATES 

C..... 


P1(K) = TSAV(I,K)+((TSAV(I+1,K)-TSAV(I,K))/

1(X(I+1)-X(I)))*(XX-X(I))


C..... 

C.....INTERPOLATE STD DEVS OF MONTHLY TEMPERATURES 

C..... 


P2(K) = VTSAV(I,K)+((VTSAV(I+1,K)-VTSAV(I,K))/

1(X(I+1)-X(I)))*(XX-X(I))

GO TO 450
 

C..... 

C.....INTERPOLATE MEAN MONTHLY RAINFALL (CM) BETWEEN YEARS

C.....OF DIFFERENT CLIMATES 

C..... 


400 P1(K) = RSAV(I,K)+((RSAV(I+1,K)-RSAV(I,K))/

1(X(I+1)-X(I)))*(XX-X(I))


C..... 

C.....INTERPOLATE STD DEVS OF MONTHLY RAINFALL 

C..... 


P2(K) = VRSAV(I,K)+((VRSAV(I+1,K)-VRSAV(I,K))/

1(X(I+1)-X(I)))*(XX-X(I))
 

450 CONTINUE
 500 CONTINUE
 

RETURN
 
END
 

C..... 
C..... 
C..... 

SUBROUTINE DISTURB 
C..... 
C.....SUBROUTINE DISTURB SIMULATES DISTURBANCES AND ALTERS TREE MORTALITY 
C.....PROBABILITIES ACCORDINGLY.  MORTALITY IS PERFORMED IN SUBROUTINE 
C.....KILL. DISTURB IS CALLED ONCE EACH YEAR BY KILL TO OBTAIN CURRENT 
C.....MORTALITY PROBABILITIES.  IT CAN BE OMITTED BY BYPASSING THE CALL 
C.....IN SUBROUTINE KILL. 
C..... 

PARAMETER (MT=350,MB=20,MS=20)
COMMON/FOREST/NCOHS(MS),DBH(MT,MB,MB),HTCM(MT,MB,MB),
 

>IAGE(MT,MB,MB),TMASS(MT,MB,MB),TLA(MT,MB,MB),
 
>LOCX(MT,MB,MB),LOCY(MT,MB,MB),ISPEC(MT,MB,MB),
 
>NOGRO(MT,MB,MB),NOGROL(MT,MB,MB),NOGROW(MT,MB,MB),
 
>NOGROT(MT,MB,MB),ICOHSIZ(MT,MB,MB),ADP(MT,MB,MB)

COMMON/PARAM/AAA(MS),DMX(MS),HMX(MS),AGEMX(MS),ITOL(MS),
 

>NTOL(MS),DRTOL(MS),G(MS),LTIME(MS),DISP(MS),DISD(MS),
 
>DREPR(MS),YLI(MS),ESTMX(MS),FC0(MS),FC1(MS),FC2(MS),
 
>HC0(MS),HC1(MS),HC2(MS),DPRBL(MS),DDMIN(MS),WITN(MS),
 
>WITX(MS),IBRW(MS),KDECID(MS)

COMMON/CONST/NSPEC,IWIDTH,KWIDTH,CELLAREA,TOTAR,NBLKS
 
COMMON/COUNT/NTOT(MB,MB),NFULL,NYEAR,KYR
 
COMMON/SEED/USEED(15)

INTEGER USEED
 

C..... 

C.....LOOP THOUGH CELLS, IDENTIFY BLOCK OF EACH CELL

C..... 


DO 85 LX=1,IWIDTH 

DO 85 LY=1,IWIDTH

IX=((LX-1)/KWIDTH)+1

IY=((LY-1)/KWIDTH)+1

IF(IX.LT.1.OR.IY.LT.1) STOP 'LOW BLOCK ERROR IN DISTURB'
 
IF(IX.GT.NBLKS.OR.IY.GT.NBLKS) 


> STOP 'HIGH BLOCK ERROR IN DISTURB' 
C..... 

C.....OPTIONS: SET ANNUAL DISTURBANCE PROBABILITY 

C..... 


DISTPR=0.005
 
IF (RAN(USEED(13)).LT.DISTPR) THEN
 

C..... 

C.....IN DISTURBED CELLS LOOP THROUGH COHORTS AND ASSIGN DEATH PROBS 

C..... 


DO 50 I=1,NTOT(IX,IY)
 
IF (LX.EQ.LOCX(I,IX,IY).AND.LY.EQ.LOCY(I,IX,IY)) ADP(I,IX,IY)=1.
 

50 CONTINUE
 
ENDIF


 85 CONTINUE
 
RETURN
 
END
 

C..... 

C.....PROGRAM REFERENCES 

C..... 

C.....BOTKIN ET AL. 1972. J. ECOL. 60:849-872. (JABOWA MODEL)

C.....BUGMANN 1994. THESIS, ZURICH. (FORCLIM MODEL) 

C.....BUGMANN 1996. ECOLOGY 77:2055-2074. (FORCLIM MODEL)

C.....BUGMANN & CRAMER 1998. FOR. ECOL. MANAGE. 103:247-263. 

C.....BUGMANN & SOLOMON 2000. ECOL. APPL. 10:95-114. (FORCLIM MODEL VERS. 2.9)

C.....BUSING 1995. J. ECOL. 83:45-53. 

C.....EMSHOFF & SISSON 1970. MACMILLAN, NEW YORK. 

C.....KER & SMITH 1955. FOR. CHRON. 31:235-246. 

C.....MOORE 1989. ECOL. MODEL. 45:63-67. 

C.....PARK & MILLER 1988. ACM 31:1192-1201 

C.....PASTOR & POST 1985. ORNL/TM-9519, OAK RIDGE. (LINKAGES MODEL) 

C.....RICHARDS 1959. J. EXP. BIOL. 10:290-300. 

C.....SHUGART & WEST 1977. J. ENVIR. MANAGE. 5:161-179. (FORET MODEL)

C.....SOLOMON 1986. OECOLOGIA 68:567-579. 

C.....THORNTHWAITE & MATHER 1957. PUBL. CLIMATOL. 10:183-311. 




Appendix B Program inputs and outputs

Input files and commands

Parameters for simulation are supplied through input files and console commands.   All files must be in fixed format.  A 
species parameter file is read by subroutine READSPEC (Appendix A).  The name of the file to be read must be specified in 
the OPEN statement in that subroutine.  In the example species file below, the first line specifies the total number of species to 
be included in the simulation species pool, followed by the identification numbers (NUM) of each species to be included.  The 
species included in a simulation can be a subset of the species in the input file.  The following lines contain parameters for each 
species.  The example file was used in the species composition and basal area tests discussed earlier (Fig. 11).  Some modifica-
tions were made to the winter temperature variables used in previous simulations in the Pacific Northwest (Busing and Solomon 
2006).  For example, new values were applied to Abies procera, Pinus monticola and Picea sitchensis.  Further evaluation of the 
parameters listed below may be required.

Example input file for species parameters:
20 1 2 3 4 5 6 7 8 91011121314151617181920

ABAM 200  75  600 1 3 0.20 340 12 0.10 0.7 20 0.08 1

0.020  6.0 -1.50 1.60 60.0 -.020 1.32 .200  390 -10   0 0  1

ABGR 225  76  300 3 3 0.45 357 12 0.10 0.7 20 0.14 1

0.020  6.0 -1.50 1.60 59.1 -.017 1.05 .300  600 -12   3 0  2 

ABLA  80  40  300 3 3 0.35 359 12 0.10 0.7 20 0.14 1

0.020  6.0 -1.50 1.60 40.0 -.030 1.23 .300  300 -99  -7 0  3   

ABPR 275  85  600 7 3 0.25 363 12 0.10 0.7 20 0.22 1

0.020 12.0 -2.30 1.40 78.6 -.013 1.19 .370  550  -7   1 0  4

ACMA 120  28  300 3 3 0.45 280 10 0.30 1.0 20 0.14 2

0.200 12.0 -2.50 1.40 30.4 -.034 0.68 .300  705  -3   7 1  5

ALRU 130  38  150 7 3 0.30 551 10 0.30 1.0 20 0.47 2

0.200 12.0 -2.50 1.40 35.6 -.028 0.80 .370  600   0   8 1  6

ARME 150  34  500 3 3 0.45 154 10 0.30 1.0 20 0.22 1

0.020 12.0 -2.30 1.40 24.2 -.034 0.89 .300  965   0   8 0  7

CHNO 370  53 3500 3 3 0.25 171 12 0.10 0.5 20 0.08 1

0.020  6.0 -1.50 1.60 45.1 -.015 1.00 .300  390 -14  -1 0  8

PIEN 244  55  600 3 3 0.40 211 12 0.20 0.7 20 0.14 1

0.200  6.0 -1.50 1.60 55.0 -.036 1.95 .300  400 -99  -3 0  9

PISI 500  90  800 3 3 0.20 374 12 0.20 0.7 20 0.14 1

0.200  6.0 -1.50 1.60 65.3 -.012 0.97 .300 1252   2   9 0 10

PICC  50  10  500 9 3 0.35 447 12 0.20 0.7 20 0.47 1

0.200  6.0 -2.90 1.70 10.0 -.016 1.14 .370 1252   3   7 0 11

PICL 213  46  600 9 3 0.35 226 12 0.20 0.7 20 0.47 1

0.200  6.0 -2.90 1.70 43.0 -.016 1.14 .370  524 -15  -9 0 12             

PIMO 200  75  600 5 3 0.30 359 12 0.20 0.7 20 0.14 1

0.020  6.0 -2.90 1.70 57.8 -.021 1.19 .370  589 -12   1 0 13    

PIPO 275  80  600 7 3 0.55 324 12 0.20 1.0 20 0.47 1

0.200  6.0 -1.50 1.60 57.4 -.013 1.11 .370  965 -12   8 0 14

PSMG 250  54  700 7 3 0.40 403 12 0.20 0.7 20 0.22 1

0.020  6.0 -2.90 1.70 55.2 -.013 1.02 .370  633 -15  -5 0 15     

PSME 425 117 1400 7 3 0.40 315 12 0.20 0.7 20 0.22 1

0.100  6.0 -2.90 1.70 84.9 -.011 0.94 .370  633 -10   5 0 16

QUGA 250  37  500 5 3 0.50 161 10 0.20 0.5 20 0.47 1

0.020 12.0 -2.30 1.40 24.8 -.027 0.92 .370  677  -4   6 0 17   

THPL 350  76 1500 3 3 0.25 282 15 0.05 0.5 20 0.08 1

0.020  6.0 -1.50 1.60 56.9 -.013 0.94 .300  748  -8   3 0 18  

TSHE 275  80  700 1 3 0.25 351 15 0.05 0.3 20 0.08 1

0.020  6.0 -1.50 1.60 66.6 -.011 0.85 .200  719  -8   4 0 19

TSME 150  46  800 1 3 0.35 203 15 0.10 0.3 20 0.08 1

0.020  6.0 -2.90 1.70 37.9 -.030 1.37 .200  300 -15  -3 0 20    

                                                       
A site file is read by subroutine READSITE and the name of the input file must be specified in the OPEN statement therein 

(Appendix A).  In the example file below, for simulations of a single site, the first line contains the number of interpolation break 
points for climate dynamics.  In this example, a site name follows; it is not read by the program.  The second line contains the 
simulation year of each break point.  The third line provides parameters for soil moisture calculations.  The following lines pro-
vide means and standard deviations of monthly temperature and precipitation (Jan.-Dec.).  The number of lines for each monthly 
parameter must be equal to the number of interpolation break points.  The example file is for simulation without long-term 
climate change, so the lines for each climate parameter contain the same values.  If climate change is to be simulated then the 
second line would contain the climate values at the second break point and so on.  If multiple sites are to be simulated in a single 
execution of the program, the parameters for subsequent sites must be appended to the file.

Example input file for site parameters:
IPOLAT= 2            BUGMANN TRANSECT SITE 1                    

     0.  1000.

PLAT=44.13 BS= 20.0 CW= 12.0 SLASP= 0.00

           3.1   4.8   6.0   7.6  10.2  13.2  15.4  15.8  14.0  10.2   6.0   3.2  Tm

           3.1   4.8   6.0   7.6  10.2  13.2  15.4  15.8  14.0  10.2   6.0   3.2  Tm

           2.0   1.7   1.8   1.9   1.3   1.2   1.0   1.6   1.1   0.9   1.7   1.8  Tsd

           2.0   1.7   1.8   1.9   1.3   1.2   1.0   1.6   1.1   0.9   1.7   1.8  Tsd

          38.7  29.8  34.6  16.4  11.1   7.4   2.6   3.9   8.0  21.4  29.8  46.4  Pm

          38.7  29.8  34.6  16.4  11.1   7.4   2.6   3.9   8.0  21.4  29.8  46.4  Pm

           3.1   3.0   2.8   1.8   1.6   1.4   1.3   1.4   1.8   2.8   2.6   3.7  Psd

           3.1   3.0   2.8   1.8   1.6   1.4   1.3   1.4   1.8   2.8   2.6   3.7  Psd

           

          

Several simulation values must be entered on the console during program execution.  They include the number of sites to 
be simulated during execution, the number of runs per site, the total number of simulation years, the output interval in years, 
and the output file name.  A prompt will appear on the console for each value.  The number of sites requires an integer entry that 
must not exceed the number of sites in the site file.  For preliminary runs try entering 1.  The number of runs per site requires an 
integer entry (from 1 to 200 is recommended).  The total number of simulation years per run requires an integer entry that does 
not exceed the greatest interpolation breakpoint value in the site file (see above).  For preliminary runs try entering 300.  The 
output interval requires an integer entry greater than or equal to 1.  For preliminary runs try entering 100.  The output file name 
requires at least one character or digit; try forel1.out for a preliminary test.

Output file

The output file is a fixed-format text file with one line per cohort of live trees in a simulation year.  It is generated by 
subroutine OUTPUT and the format can be obtained by examining the FORMAT statement therein (Appendix A).  A portion 
of an example output file is displayed below.  From left to right the variables are listed for site (IS), run (IRUN), simulation 
year (KYR), species (ISPEC), cohort size (ICOHSIZ), tree diameter at breast height (DBH, cm), tree height (HTM, m), cohort 
leaf area (TLA), cohort age (IAGE, yr), cell X coordinate (LOCX), cell Y coordinate (LOCY), consecutive years of suppressed 
growth (NOGRO), consecutive years of light suppression (NOGROL), consecutive years of moisture suppression (NOGROW), 
consecutive years of temperature suppression (NOGROT), annual drought for evergreen species (DRE), annual drought for 
deciduous species (DRD), annual degree-days for evergreen species (DDE), annual degree-days for deciduous species (DDD), 
minimum winter temperature (TCOLD, oC), and simulation tract area (TOTAR, ha).

Example of an output file portion:
  1  1    1   7   4   1.6   0.0     0.0    1   1   1  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  19   4   2.0   0.0     0.0    1   1   1  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1   4   1   2.1   0.0     0.0    1   1   2  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1   7   3   1.6   0.0     0.0    1   2   1  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1   4   4   2.1   0.0     0.0    1   2   2  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  14   2   2.0   0.0     0.0    1   2   3  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1   5   2   1.9   0.0     0.0    1   3   1  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  16   2   2.0   0.0     0.0    1   3   1  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  13   4   2.0   0.0     0.0    1   3   2  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  17   1   1.7   0.0     0.0    1   3   2  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  13   1   2.0   0.0     0.0    1   3   3  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  18   5   1.9   0.0     0.0    1   3   3  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1  18   1   1.9   0.0     0.0    1   1   5  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

  1  1    1   4   4   2.1   0.0     0.0    1   2   5  0  0  0  0 0.08 0.11  1894.8  1642.3   0.6  13.0

